

Simulace v OrCadPSPice 10 na úrovni vstupních souborů (bez schématického editoru)

Spouštěný program: PSpice.exe (ikona PSpice AD Demo), nikoliv Capture CIS Demo.

Základní metody tvorby vstupního souboru:


A) Tvorba vstupního souboru externím programem.

1. V jakémkoliv textovém manažeru, např. v NotePadu, vytvoříme vstupní soubor pro simulaci. Jeho přípona bude cir.
2. V prostředí jakéhokoliv „file manageru“ (Windows Commander, Průzkumník..) kliknutím na soubor spustíme PSpice (pokud jsou soubory s příponou „cir“ asociovány s PSpicem). Nebo použijeme jiný postup, viz 3:
3. Spustíme PSpice.exe, zvolíme „File/Open simulation“, v okně „Open Simulation“ nastavíme masku „Soubory typu“ na „Circuit Files (*.cir)“ a vybereme náš vstupní soubor.

B) Tvorba vstupního souboru přímo v PSpice

1. Spustíme PSpice.exe, zvolíme „File/New/Text File“ (alternativou je horká klávesa Ctrl N).
2. Vytvoříme vstupní soubor a uložíme jej volbou „File/Save As“. Masku typu souboru může být nastavena jakákoliv, avšak název souboru musíme napsat úplný, t.j. s příponou cir!!!
3. Zvolíme „File/Open simulation“, v okně „Open Simulation“ nastavíme masku „Soubory typu“ na „Circuit Files (*.cir)“ a vybereme náš vstupní soubor.

Připravenost programu k simulaci na základě vstupního souboru je indikována modrou šipkou na horní liště (ikona pro spouštění simulace).



Práce s několika soubory:

Často potřebujeme pracovat s několika textovými soubory „současně“. Například tvoříme vstupní soubor pro simulaci a potřebujeme založit svou vlastní novou knihovnu nebo editovat jinou knihovnu na disku.

V prostředí PSpice není možné mít současně otevřených více vstupních souborů typu *.cir. Pokus o otevření dalšího vstupního souboru vede na deaktivaci souboru předchozího.


Další soubor jiného typu než vstupní, např. knihovnu, otevřeme volbou „File/Open“. Zvolíme příslušnou masku (v případě nutnosti „All Files“) a vybereme daný soubor. Chceme-li založit nový soubor, který na disku ještě není, provedeme volbu „File/New/Text File“ a po jeho vytvoření jej uložíme pod úplným názvem, tedy i s příponou.

Mezi otevřenými soubory přecházíme klikáním na jejich záložky podobně jako na listy v Excelu. PSpice sice nezobrazuje přípony v názvech souborů, avšak vstupní soubor poznáme podle toho, že je za ním připojena poznámka „(active)“. Kromě toho je název „aktivní úlohy“, která je připravena k simulaci, napsán v okénku na horní liště (viz název „RC“ na obrázku na předchozí straně).

Simulace:

Aktivuje se kliknutím na “modrou šipku” (viz obrázek na předchozí straně) nebo volbou “Simulation/Run”.

Pokud jsou ve vstupním souboru instrukce pro výstup dat např. do “Probe”, otevře se příslušný další list. Příslušné výstupy simulace je možno zobrazovat a volit pomocí levého sloupce ikon:



Při práci se simulační úlohou často pracujeme tak, že po prozkoumání výsledků simulace, chybových hlášení atd. se vracíme do okna vstupního souboru či do okna knihovny, provedeme určité úpravy a poté opět spouštíme simulaci. Před spuštěním simulace se nás program zeptá, zda chceme uložit provedené změny. Proto je vhodné vždy před kliknutím na “modrou šipku” kliknout na ikonu s disketou.


Struktura vstupního souboru:

VSTUPNÍ SOUBOR (*.CIR)

Hlavička
Popis obvodu = netlist
Příkazy pro řízení simulace
.end

PŘÍKLAD: RLC.CIR

RLC obvod
* Vzdroj 1 0 4.5V Ls 1 2 10mH Rs 2 3 100 Rp 3 0 3.3k Cp 3 0 220nF *
.TRAN 1u 1m SKIPBP .PROBE V(Cp) V(Ls) *
.end



Sbírka příkladů na vstupní soubory

Zásady práce se sbírkou:

1. Výchoziskem práce je „papírové“ zadání schématu obvodu a požadavky na cíle analýzy.
2. Prvním cílem práce je napsání vstupního souboru. Student může mít k dispozici všechny dostupné materiály a literaturu bez jakéhokoliv omezení, během cvičení i vydatnou pomoc učitele.
3. Při použití nového příkazu, funkce apod. by si student měl tyto techniky fixovat jako vzory možného řešení jiných konkrétních problémů počítačové simulace. Dané syntaxe nemusí zpočátku umět z paměti, měl by však vědět, kde hledat, až je bude později potřebovat. Měl by si později v manuálech nalézt další varianty a možnosti daných příkazů a orientačně se s nimi seznámit.
3. Preferovanou metodou při práci se SPICE je metoda pokusu a omylu ve smyčce vstupní soubor-simulace, která by měla postupně konvergovat k bezchybným výsledkům. Důležité je, aby student přicházel na řešení alespoň některých problémů sám. Neméně důležité je, aby se zpětně naučil lépe rozumět obvodům, které simuluje. Pro vyučujícího to znamená dvě věci: Dát studentům přiměřený návod a pak i čas na samostatné řešení, pak předvést správné postupy, a v závěru příkladu komentovat výsledky simulací **jako elektrikář**. Pro studenty to znamená mít po ruce vhodné přehledy syntaxí jazyka SPICE, jakož i zápisky z přednášek, a schopnost provádět efektivní záznamy o probíhajících počítačových experimentech do pracovních sešitů. Důležité je i samostatné procházení příkladů mimo organizovanou výuku na vlastním počítači.

Příklady 1 až 16

Základy tvorby vstupních souborů (*.cir)

Základy modelování pasivních prvků (R, C..) a stejnosměrných zdrojů napětí (V).

Příkaz .OP.

Způsob používání příkazů .PARAM a .STEP.

Příkaz .TEMP.

Práce s diodami (D) a tranzistory (Q). Základy práce s příkazem .MODEL.

Příkaz .LIB.

1)

Odporovy delic

*


R1 1 2 4k; rezistor mezi uzly 1 a 2, odpor 4kohm

R2 2 0 1k; rezistor mezi uzly 2 a 0, odpor 1kohm

V1 1 0 DC 10V; zdroj stejnosmerneho napeti 10V mezi uzly 1 a 0

*

.END



Není specifikována žádná analýza. Přesto proběhne výpočet ss pracovního bodu (implicitně proběhne vždy) s výsledkem

```
NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE  NODE VOLTAGE
( 1) 10.0000 ( 2)  2.0000
```

Vyzkoumejte, co značí údaj TOTAL POWER DISSIPATION, ověřte vypočítanou hodnotu.

Vyzkoumejte, co značí údaj VOLTAGE SOURCE CURRENTS, ověřte vypočítanou hodnotu. Jak je to se znaménkem?

Náměty k experimentům:

1) Vynechejte u zdroje V1 výraz DC. Výsledek bude stejný. Proč? (syntaxe!!!)

2) U zdroje V1 zaměňte pořadí uzlů, tedy 0 1. Výsledná napětí budou záporná. Proč?

3) Přidejte před poslední příkaz .END příkaz .OP. Ve výsledku není žádná změna. Proč?

4) Vyzkoušejte různé jiné názvy než R1, R2, V1. Vyzkoušejte si, že důležité je neměnit první identifikační znak ve jméně a že to funguje i bez dalších znaků, tedy že jsou povolena i jména R a V.

2)

Doplňte vstupní soubor takto:

Odporovy delic

*

R1 1 2 {Rpromenny}

R2 2 0 1k

V1 1 0 10V

*

.PARAM Rpromenny=4k

*

.END

Výsledek je pořád stejný. Vysvětlení o co tady jde (globální parametry simulace, výrazy v závorkách {}).

Doplníme takto:

3)

Odporovy delic

*

R1 1 2 {Rpromenny}

R2 2 0 1k

V1 1 0 10V

*


```
.PARAM Rpromenny=4k
.STEP param=Rpromenny LIST 1k 2k 4k
*
.END
```

Vysvětlení, co se stalo (krokování odporu Rpromenny, opakovaný výpočet pracovního bodu).
Ověření výsledků.

Připomenutí dalších metod krokování, resp. odkaz na minimanuál.

Tip: zkuste do seznamu krokovaných hodnot zařadit odpor 0. Není povoleno!

Modifikujte vstupní soubor tak, aby modeloval obvod s diodou:



4)

Obvod s diodou

*

R1 1 2 4k; rezistor mezi uzly 1 a 2, odpor 4kohmy

D1 2 0 obycejna_dioda; dioda mezi uzly 2 a 0, jmenuje se obycejna_dioda, anodu ma na uzlu 2, +katodu na uzlu 0

V1 1 0 10V; zdroj stejnosmerneho napeti 10V mezi uzly 1 a 0

*

.model obycejna_dioda D; diode obycejna_dioda je prirazen implicitni model diody D

*

.END

Diskuse nad příkazem .MODEL. Připomenutí teorie.

Diskuse nad výsledky simulace.

Přidejte nad příkaz .END příkaz .OP. Co se stane? Vysvětlení.

Přehod'te pořadí vývodů diody a potvrďte analýzou to, co si myslíte.

Modifikujte soubor takto:

5)

Obvod s diodou

*

R1 1 2 4k

D1 2 0 obycejna_dioda

V1 1 0 10V

*

.model obycejna_dioda D

*

.TEMP 50

.OP

*

.END

Teplota je nyní 50 stupňů Celsia, ne implicitně 27, změnilo se napětí na diodě?

6)

Obvod s diodou

*

R1 1 2 4k

D1 2 0 obycejna_dioda

V1 1 0 10V

.model obycejna_dioda D

.TEMP 27 50 100

*

.END

Krokování teploty, příkaz .OP je vynechán.

7)

Obvod s diodou

*

R1 1 2 4k

D1 2 0 obycejna_dioda

V1 1 0 10V

.model obycejna_dioda **D(Is=80p Rs=2)**

*

.END

Připomenutí významu tzv. implicitních parametrů modelu.

8)

Stabilizator se Zenerovou diodou

*

R1 1 2 **250**; rezistor mezi uzly 1 a 2, odpor 250ohmuD1 **0 2 D1N750**; Zenerova dioda polovana v zavernem smeru, Zenerovo napeti je asi 4.7V pri proudu 20mA


V1 1 0 10V; zdroj stejnosmerneho napeti 10V mezi uzly 1 a 0

*

.model D1N750 D(Is=880.5E-18 Rs=.25 Ikf=0 N=1 Xti=3 Eg=1.11 Cjo=175p M=.5516**+ Vj=.75 Fc=.5 Isr=1.859n Nr=2 Bv=4.7 Ibv=20.245m Nbv=1.6989****+ Ibvl=1.9556m Nbv1=14.976 Tbv1=-21.277u)**

*

.END



Příslušný model prosím „ukrást“ přes schránku z knihovny EVAL.LIB, příp. z jiné (u profi verze OrCadPSpice je v knihovně DIODE.LIB).

Diskuse.

9)

Stabilizator se Zenerovou diodou

*

R1 1 2 250;

D1 0 2 D1N750

V1 1 0 10V

*

.lib eval.lib

*

.END

Místo **eval** případně napíšeme správný název knihovny s diodou D1N750. Diskuse o významu knihoven.

10)

Zjistěte teplotní součinitel výstupního napětí: vypočtete stabilizované napětí při teplotě 27 stupňů (V27) a pak 37 stupňů (V37), součinitel= $(V37-V27)/V27/10$. Výsledek: $-23,4 \cdot 10^{-6}$ /stupeň.

11)

Zjistěte činitel stabilizace $\Delta V(1)/\Delta V(2)$. Návod - Vypočtete výstupní napětí při vstupním napětí 10V a pak při 20V. Můžete provést opakovaný výpočet nebo využít příkaz **.STEP**.

Výsledek: 169,8.

12)

zesilovac

*

Vin 1 0 AC 1 sin 0 10mV 1kHz

Cv1 1 2 10uF

Rb1 6 2 16.7k

Rb2 2 0 3.7k

QT 3 2 5 obycejny_tranzistor; pořadí vývodů tranzistoru: kolektor, báze, emitor

Rc 6 3 1k

Re 5 0 200

Cv2 3 4 10u

Rz 4 0 5k

Vbat 6 0 10V


*

.model obycejny_tranzistor NPN; NPN je jméno modelu bipolárního tranzistoru

.OP

*

.end



Vysvětlete význam řádku s definicí Vin. U modelu tranzistoru jsou uvažovány implicitní parametry. Opište si souřadnice vypočteného pracovního bodu, parametry BF, betaDC, betaAC tranzistoru. Poznatek: BF (Beta Forward) není přesně rovno „betě“ tranzistoru, od tohoto čísla se však proudové zesilovací činitele odvíjejí.

Pomocí odečteného parametru betaDC ověřte výpočtem daný pracovní bod.

Všimněte si nezvykle velkého napětí mezi bází a editorem. Ne všechny implicitní hodnoty parametrů tranzistoru jsou „optimální“ pro konkrétní aplikaci. Pro konkrétní aplikaci se vyplatí použít model konkrétní součástky – viz dále.

13)

Jak donutit SPICE, aby vypsal proud rezistorem Rb2? SPICE vypisuje proudy procházející zdroji napětí. Proto umístíme do série s Rb2 zdroj V, ale jeho napětí nastavíme na 0V, aby nebyla ovlivněna funkce zesilovače.

Výsledek: $I(Rb2)=0,4661mA$.

14)

Zjistěte změny pracovního bodu po výměně tranzistoru za jiný s parametrem BF 100, 200 a 500.
Řešení:

zesilovac

*

Vin 1 0 AC 1 sin 0 10mV 1kHz

Cv1 1 2 10uF

Rb1 6 2 16.7k

Rb2 2 0 3.7k

QT 3 2 5 obycejny_tranzistor

Rc 6 3 1k

Re 5 0 200

Cv2 3 4 10u

Rz 4 0 5k

Vbat 6 0 10V

*

.param BF=100

.model obycejny_tranzistor NPN BF={BF}

.step param=BF list 100 200 500

.OP

*

.end

Všimněte si, že parametry modelu nemusí být v závorkách, neboli že jsou přípustné oba zápisy:

.model obycejny_tranzistor NPN BF={BF}

.model obycejny_tranzistor NPN (BF={BF})

15)

Zjistěte stabilitu pracovního bodu při změnách teploty. Konkrétně pro BF=100 „změřte“ klidový kolektorový proud pro teplotu 20 stupňů (I_{c20}) a 30 stupňů (I_{c30}) a pak vypočtete činitel teplotní stabilizace $100 \cdot (I_{c30} - I_{c20}) / (10 \cdot I_{c20})$, tj. procentuální změnu I_c při změně teploty o 1 stupeň.

Výsledek: Při vzrůstu teploty o 1 stupeň vzroste klidový proud kolektoru asi o 0,12 procenta.

16)

Najděte v knihovnách model tranzistoru Q2N2222 a nahraďte jím původní implicitní model. Buď si model překopírujte do vašeho vstupního souboru nebo se napojte na knihovnu příkazem **.lib**.

V ukázce je zvolena první možnost:

zesilovac

*

Vin 1 0 AC 1 sin 0 10mV 1kHz

Cv1 1 2 10uF

Rb1 6 2 16.7k

Rb2 2 0 3.7k

QT 3 2 5 **Q2N2222**

Rc 6 3 1k
Re 5 0 200
Cv2 3 4 10u
Rz 4 0 5k
Vbat 6 0 10V
*

.model Q2N2222 NPN(Is=14.34f Xti=3 Eg=1.11 Vaf=74.03 Bf=255.9 Ne=1.307
+ Ise=14.34f Ikf=.2847 Xtb=1.5 Br=6.092 Nc=2 Isc=0 Ikr=0 Rc=1
+ Cjc=7.306p Mjc=.3416 Vjc=.75 Fc=.5 Cje=22.01p Mje=.377 Vje=.75
+ Tr=46.91n Tf=411.1p Itf=.6 Vtf=1.7 Xtf=3 Rb=10)
*
.OP
*
.end

Ověřte, že klidové napětí báze-emitor je nyní 0,688V.

Doporučujeme zopakovat s tímto tranzistorem experimenty z příkladu **12**.

Příklady 17 až 26

Základy práce s podobvody SPICE (.SUBCKT)

Zakládání vlastních knihoven.

Zdroje napětí typu E.

Užitečné funkce LIMIT a TABLE.

Definování uživatelských funkcí - .FUNC.

17)

obvod s operacním zesilovacem

*

Vin 1 0 1V

R1 1 2 10k

R2 2 3 10k

R3 3 0 1k

R4 3 4 10k

X1 0 2 4 operak_jednoduchy

*

.subckt operak_jednoduchy 1 2 4

Rin 1 2 1meg

Rout 3 4 50

E1 3 0 1 2 200k



.ends

*

.OP

*

.end



Na základě schématu podobvodu a schématu celého obvodu se snažte pochopit mechanismus tohoto hierarchického modelování.

Zopakujte si z teorie způsob modelování zdroje napětí řízeného napětím E.

Proveďte analýzu pracovního bodu a ručním výpočtem ověřte její správnost. Zdůvodněte nepatrné numerické odchylky.

Všimněte si výrazu

(X1.3) -12.0540

(interní uzel č. 3 podobvodu X1 a jeho napětí; tento uzel není připojen k vnějším uzlům obvodu, proto jeho napětí SPICE rovněž zviditelní).

18)

obvod s operacním zesilovacem

*

Vin 1 0 1V

R1 1 2 10k

R2 2 3 10k

R3 3 0 1k

R4 3 4 10k

X1 0 2 4 operak_jednoduchy **params: A=1T Rin=1e12 Rout=1e-10**

*

.subckt operak_jednoduchy 1 2 4 **params: A=200k Rin=1meg Rout=50**Rin 1 2 {**Rin**}Rout 3 4 {**Rout**}E1 3 0 1 2 {**A**}

.ends

*

.OP

*

.end

Nyní je podobvod volán s parametry, a to tak, abychom simulovali „téměř ideální“ operační zesilovač. Poznamenejme, že Rout nelze nastavit na nulu a žádný číselný údaj na nekonečno.

Volání podobvodu s parametry je zvlášť výhodné, pokud je podobvod umístěn v knihovně. Pak jej může využívat více uživatelů ve svých konkrétních vstupních souborech, které jsou na tuto knihovnu napojeny příkazem .lib.

19)

Zkuste si vytvořit svou vlastní knihovnu se jménem pokus.lib, do které vložíte definiční text podobvodu:

vstupní soubor:

obvod s operacním zesilovcem

*

Vin 1 0 1V

R1 1 2 10k

R2 2 3 10k

R3 3 0 1k

R4 3 4 10k

X1 0 2 4 operak_jednoduchy **params: A=1T Rin=1e12 Rout=1e-10**

*

.lib pokus.lib

.OP

*

.end

Knihovna pokus.lib:

Moje pokusna knihovna

*

.subckt operak_jednoduchy 1 2 4 **params: A=200k Rin=1meg Rout=50**

Rin 1 2 **{Rin}**

Rout 3 4 **{Rout}**

E1 3 0 1 2 **{A}**

.ends

20)

Knihovnu pokus.lib rozšíříme o model operačního zesilovače se saturací. Výstupní napětí operačního zesilovače pak nemůže být větší než zvolená hodnota 13V a menší než -13V. Je k tomu využít speciální zápis zdroje E spolu s funkcí LIMIT (viz výpis níže). Prostudujte!

obvod s operacním zesilovcem

*

Vin 1 0 1V

R1 1 2 10k

R2 2 3 10k

R3 3 0 1k

R4 3 4 10k

X1 0 2 4 **operak_se_saturaci** **params: A=200k Rin=1meg Rout=50**

*

.lib pokus.lib

.OP

*

.end

moje pokusna knihovna

*

```
.subckt operak_jednoduchy 1 2 4 params: A=200k Rin=1meg Rout=50
```

```
Rin 1 2 {Rin}
```

```
Rout 3 4 {Rout}
```

```
E1 3 0 1 2 {A}
```

```
.ends
```

```
.subckt operak_se_saturaci 1 2 4 params: A=200k Rin=1meg Rout=50
```

```
Rin 1 2 {Rin}
```

```
Rout 3 4 {Rout}
```

```
E1 3 0 value={limit(A*V(1,2), -13,13)}
```

```
.ends
```

Analýzou se přesvědčete o tom, že naprogramování saturace funguje (například zvětšíte vstupní napětí na 20V nebo zmenšíte R3 na 100 ohmů).

21)

Pokuste se napětíovou převodní charakteristiku operačního zesilovače modelovat funkcí TABLE.

Řešení:

```
E1 3 0 value={table(V(1,2), -13/A, -13, 13/A,13)}
```

22)

Pomocí příkazu .FUNC definujte novou funkci y=OUTPUT(x), která bude modelovat závislost výstupního napětí zdroje E1, tj. y, na vstupním diferenčním napětí, tj. x. Tuto funkci pak použijte v podobvodu operak_se_saturaci.

Řešení – podobvod:

```
.subckt operak_se_saturaci 1 2 4 params: A=200k Rin=1meg Rout=50
```

```
Rin 1 2 {Rin}
```

```
Rout 3 4 {Rout}
```

```
E1 3 0 value={output(V(1,2))}
```

```
.FUNC output(x)=limit({A}*x, -13,13); funguje i bez rovnítka a taky {limit(A*x, -13,13)}
```

```
.ends ; dokonce nemusí být funkce limit ani v { }
```

Funkce má pouze lokální platnost v rámci podobvodu.

Zkuste změnit parametry operačního zesilovače tak, aby představoval ideální OZ, viz příklad 15. SPICE bude mít problémy s nalezením řešení. O této problematice bude pojednáno později.

23)

Námět na další zdokonalení modelu: napište podobvod operačního zesilovače, který bude mít další dva vývody na napájecí napětí, a saturační napětí bude odvozeno z napájecího vzorcem

Saturační napětí = napájecí napětí -1,5V. Řešení:


obvod s operacnim zesilovcem

*

```
Vin 1 0 1V
```

```
Vplus 5 0 15V
```

```
Vminus 6 0 -15V
```



```

R1 1 2 10k
R2 2 3 10k
R3 3 0 1k
R4 3 4 10k
X1 0 2 5 6 4 operak_se_saturacix params: A=200k Rin=1meg Rout=50
*

```

```

.lib pokus.lib
*

```

```

.OP
*

```

```

.end

```

```


* uzly:                neinvertující vstup
*                      | invertující vstup
*                      | | kladné napájecí napětí
*                      | | | záporné napájecí napětí
*                      | | | | výstup
*                      | | | | |

```

```

.subckt operak_se_saturacix 1 2 3 4 5 params: A=200k Rin=1meg Rout=50
Rin 1 2 {Rin}
Rout 5 6 {Rout}
R1 3 0 1T
R2 4 0 1T
E1 6 0 value={limit(A*V(1,2),V(4)+1.5,V(3)-1.5)}
.ends

```



Pomocné odpory 1Tohm (Terra = $1 \cdot 10^{12}$) jsou připojeny k napájecím vývodům z toho důvodu, že každý z uzlů obvodu musí mít stejnosměrné spojení se zemí (s uzlem 0).
Odzkoušejte nyní fungování celého obvodu v saturačním režimu při různých napájecích napětích.

24)

Nahraďte model OZ profesionálním SPICE modelem zesilovače typu 741 z knihovny PSPICE.
Řešení – viz předchozí příklad, pouze volání podobvodu bude jiné:

```

X1 0 2 5 6 4 uA741

```

V demo verzi OrCadPSpice je tento model v knihovně EVAL.LIB, v profi verzi je v knihovně OPAMP.LIB. Prohlédněte si jej. Pořadí vývodů operačního zesilovače při volání podobvodu je standardně (viz též hlavička podobvodu v knihovně): neinvertující vstup, invertující vstup, kladné napájení, záporné napájení, výstup.

Pro úspěšnou simulaci potřebujeme tuto informaci, nepotřebujeme rozumět „vnitřku“ daného podobvodu.

Porovnáme-li chování obvodu s operačním zesilovačem 741 a s naším zesilovačem operak_se_saturacix, nezaznamenané velké rozdíly. Je tomu tak proto, že náš model je dostatečně přesný pro analýzu stejnosměrných poměrů (při napájecím napětí nad 1,5V). Naprosto nevyhovující však bude pro analýzu v dynamickém režimu (řešení přechodných dějů a kmitočtových charakteristik).

25)

Z Internetu nebo odjinud si sežňte SPICE model integrované napěťové reference REF01 firmy Analog Devices. Příslušný podobvod si uložte do pomocné knihovny POKUS.LIB. Zjistěte výstupní napětí podle schématu na obrázku při teplotě 27 stupňů Celsia.

Řešení – vstupní soubor:

zdroj 10V

*

Vnap 1 0 15V

R1 2 3 5k

R2 3 0 5k

Xreference 1 0 3 2 REF01_AD

*


.lib pokus.lib

*

.end

Výstupní napětí vyjde 9,0977V.

Diskuse: zjišťování vývodů obvodu z hlavičky příslušného podobvodu, souvislost se zápisem podobvodu ve vstupním souboru.



```
*
*          VIN
*          | GND
*          | | TRIM
*          | | | VOUT
*          | | |
.SUBCKT REF01_AD 2 4 5 6
.....
.....
.ENDS REF01_AD
```

Náměty na další pokusy, viz příklad 26:

26)

Pomocí příkazu .PARAM realizujte odpory R1 a R2 jako potenciometr o odporu 10kohmů. Zjistěte, jak se mění výstupní napětí při „točení“ potenciometru. Pokuste se tak nastavit výstupní napětí přesně na 10V.

Řešení:

zdroj 10V

*

Vnap 1 0 15V

R1 2 3 {Rprom}

R2 3 0 {10k-Rprom}

Xreference 1 0 3 2 REF01_AD

*

.param Rprom=5k

.lib pokus.lib

.step param=Rprom list 8k 8.5k 9k

*

.end

Při růstu R1 roste výstupní napětí. Napětí 10V je nastaveno při R1 asi 8,77k.

Příklady 27 až 48

Příkaz .NODESET

Práce s řízenými zdroji včetně polynomiálních

Stejnoseměrná analýza – příkaz .DC. Základní a parametrický režim (jednoduché a vnořené rozmítání).

Příkaz .PROBE

Příkaz .PRINT

Zobecněná DC analýza (rozmítání jiných proměnných než napětí a proudů).

27)

Nalezněte všechny stejnosměrné pracovní body bistabilního klopného obvodu s tranzistory podle obrázku.

Řešení:

flipflop

Vbat 1 0 10V

R1 1 2 1k

R2 1 3 1k

R3 2 5 56k


R4 3 4 56k

Q1 2 4 0 Q2N2221

Q2 3 5 0 Q2N2221

.lib bipolar.lib

.end



PSpice nalezne tzv. nestabilní stejnosměrný pracovní bod, kdy oba tranzistory jsou v aktivním režimu (v praxi se nemůže udržet): $V(2) = V(3) = 4.7792V$.

K nalezení stabilního pracovního bodu, odpovídajícího otevřenému Q1 a uzavřenému Q2, přidáme příkaz .NODESET, např.

.nodeset V(3)=10V

Pak vyjde $V(2)=0.1244V$, $V(3)=9.8370V$

Vyzkoušejte obdobně nastavit pracovní bod pro uzavřený Q1 a otevřený Q2.


28)

Nalezněte všechny stejnosměrné pracovní body invertujícího komparátoru s hysterezí.

V praxi může nastat pouze saturace OZ, neboť v obvodu působí kladná zpětná vazba. Kladné saturaci odpovídá napětí $V(2)$ cca +14V, záporné saturaci napětí $V(2)$ cca -14V.

V obvodu může nastat další teoretický stav – nestabilní rovnováha, kdy diferenční napětí je nulové, z toho plynoucí napětí $V(3)=1V$ a $V(4)=2V$. Tento stav, který je běžně pozorovatelný u záporné zpětné vazby (při přehozených vývodech OZ), je však v tomto zapojení trvale neudržitelný.

Vyzkoušejte, který pracovní bod nalezne PSpice bez příkazu .NODESET. Pak nalezněte všechny stavy.



invert.komparator s OZ

Vin 1 0 1V

R1 2 3 1k

R2 3 0 1k

X1 3 1 5 4 2 LF411

Vplus 5 0 15V

Vminus 0 4 15V

.lib

.end

Bez příkazu .NODESET je nalezen nestabilní pracovní bod (!).

S příkazy .NODESET V(2)=15V, případně .NODESET V(2)=-15V, jsou nalezeny stabilní pracovní body, kdy V(2)=14.2960V, případně V(2)=-14.2960V.


29)

Přehod'te vstupní svorky OZ z předchozího příkladu, takže dostanete lineární zesilovač se zesílením 2. Zjistěte výstupní napětí. Zjistěte, zda je možné příkazem .NODESET nalézt další řešení.

Výsledky: Výstupní napětí bude nyní 2V, žádné další řešení neexistuje (jde o lineární úlohu s jediným řešením, zpětná vazba je záporná, obvod je stabilní).

30)

Vypoč'te napětí na výstupu kvadrátoru. Zjistěte, zda je možné příkazem .NODESET nalézt další řešení.



Obvod má dvě stejnosměrná řešení: V(3)=250mV a V(3)=2.25V. Ověřte si to.

obvod s kvadrátorem

Vin 1 0 750mV

R1 1 2 1k

R2 2 3 1k

Ekvad 3 0 VALUE={V(2)^2}

.end

PSpice nalezne řešení V(3)=250mV. Další řešení nalezne např. po použití příkazu .NODESET V(2)=2V. Vyzkoušejte, že nastavování V(3) nemá efekt (jedná se o výstupní napětí řízeného zdroje).

31)

Modelujte kvadrátor z příkladu 30 polynomiálním zdrojem.


Řešení: Ekvad 3 0 POLY(1) 2 0 0 0 1

32)

Pokuste se nalézt výstupní napětí z příkladu 30, je-li vstupní napětí V(1)=1V.

Obvod má nyní jediné řešení V(3)=1V. Toto řešení je velmi citlivé na parametry součástek. Při vstupním napětí větším než 1V dokonce řešení (v rámci tohoto modelu obvodu) neexistuje!

PSpice bude mít nyní problémy s konvergencí řešení. Odpomoc – modifikace globálních podmínek simulace. Změníme parametr ABSTOL ze standardního 10^{-12} na 10^{-10} a potvrdíme „OK & resume simulation“. Podrobnosti později. Částečně si lze pomoci i příkazem .NODESET, ale musíme dobře odhadnout konečné řešení, např. .NODESET V(2)=0.9V.




33)

Zjistěte stejnosměrnou napět'ovou převodní charakteristiku zesilovače, tj. graf závislosti výstupního napětí na vstupním

napětí, v rozsahu vstupního napětí od -10V do +10V. Z charakteristiky odvoďte střídavé zesílení, saturační úrovně a rozsah vstupního napětí, v němž se zesilovač chová jako lineární prvek.

```
Vin 1 0 1V
R1 2 3 1k
R2 3 0 1k
Xoz 1 3 4 5 2 LF411
Vplus 4 0 15V
Vminus 5 0 -15V
.dc Vin -10 10 0.01 ; rozmítání Vin od -10V do +10V s krokem 0.01V
.probe ; vygenerování souboru *.dat pro postprocesor PROBE a spuštění PROBE
.lib
.end
```

Po otevření prázdného okna PROBE přidáme křivku (Add Trace), odpovídající napětí V(2).



Výsledky: saturační úrovně -14.296V, +14.296V, hranice lineárního režimu pro V(1) jsou -7.15V a +7.15V (lze změřit pomocí kurzorů – „Toggle Cursor“, jemně krokovat po vypočtených bodech „Cursor Point“. Pak střídavé zesílení vychází cca $14.296/7.15=1.9994$. Podle teorie má vyjít 2. Zesílení lze i automaticky změřit: Trace/Measurement, vybereme SlewRate_Rise, Eval, Name of trace to search: V(2). OK. Objeví se výsledek 1.99998 (rozdíl oproti předchozímu výpočtu je způsoben zaokrouhlováním mezivýsledků).

Další možnost automatického měření: Trace/Evaluate measurement...

Poznámka: Zkuste nahradit příkaz .PROBE příkazem .PROBE V(2). Význam – viz přednášky.

34)

Získejte charakteristiky zesilovače z příkladu 33 pro odpory R1 z řady 100ohmů, 1kohm, 10kohmů.

neinvertující zesilovač s OZ

```
Vin 1 0 1V
R1 2 3 {R1}
R2 3 0 1k
Xoz 1 3 4 5 2 LF411
```

```
Vplus 4 0 15V
Vminus 5 0 -15V
.dc Vin -10 10 0.01
.param R1 1k
.step param R1 list 100 1k 10k
.probe
.lib
.end
```

35)

Získejte charakteristiky zesilovače z příkladu 33 pro symetrické napájecí napětí (10, 11, 12, 13, 14, 15)V.

neinvertující zesilovác s OZ

```
Vin 1 0 1V
R1 2 3 1k
R2 3 0 1k
Xoz 1 3 4 5 2 LF411
Vplus 4 0 {Vnap}
Vminus 5 0 {-Vnap}
.dc Vin -10 10 0.01
.param Vnap 15V
.step param Vnap 10 15 1; lineární krokování od 10V do 15V s krokem 1
.probe
.lib
.end
```

Rozbor, jak souvisí napájecí napětí se saturačním napětím.

36)

Vyřešte předchozí příklad s využitím vnořeného rozmítání v příkazu .DC.

neinvertující zesilovác s OZ

```
Vin 1 0 1V
R1 2 3 1k
R2 3 0 1k
Xoz 1 3 4 5 2 LF411
Vplus 4 0 {Vnap}
Vminus 5 0 {-Vnap}
.dc Vin -10 10 0.01 param Vnap 10 15 1
.param Vnap 15V
.probe
.lib
.end
```

37)

Nakreslete stejnosměrnou převodní charakteristiku komparátoru s hysterezí z příkladu 28 při změně vstupního napětí a) z -10V do +10V, b) z +10V do -10V.

a)

invert.komparator s OZ

Vin 1 0 1V

R1 2 3 1k

R2 3 0 1k

X1 3 1 5 4 2 LF411

Vplus 5 0 15V


Vminus 4 0 -15V

.dc Vin -10 10 0.01

.probe

.lib


.end



b)

Modifikujeme příkaz .DC:

.DC Vin 10 -10 0.01



Z výsledků simulací je vidět hystereze. Příkaz .DC nemůže být použit vícekrát v rámci jednoho vstupního souboru, takže běžně nelze při stejnosměrné analýze vykreslit hysterezní charakteristiku. Překlápěcí úrovně vstupního napětí jsou rovny polovině hysterezního napětí, tedy asi $\pm 7.14\text{V}$.

38)

Nakreslete stejnosměrnou převodní charakteristiku komparátoru s hysterezí z příkladu 28 při změně vstupního napětí a) z -7V do +7V, b) z +7V do -7V.

invert.komparator s OZ

Vin 1 0 1V

R1 2 3 1k

R2 3 0 1k

X1 3 1 5 4 2 LF411

Vplus 5 0 15V

Vminus 4 0 -15V


.dc Vin -7 7 0.01; v dalším pokusu změníme na .dc Vin 7 -7 0.01

.probe

.lib

.end

Tentokrát PSpice vypočte v obou případech nesprávnou charakteristiku:



Na vině je nalezení nestabilního (tedy nesprávného) pracovního bodu hned při výpočtu prvního bodu křivky. Při vstupním napětí -7V, resp. +7V jsou totiž tři možná řešení. Při vstupním napětí -10V, resp. +10V (předchozí příklady) je jediné možné řešení – saturace. Obrázek tak vlastně ukazuje část převodní charakteristiky jiného obvodu – lineárního zesilovače se zesílením 2 z příkladu 33.

Vyřešení problému – kombinace s příkazem .NODESET:

.NODESET V(2)=15V pro a), .NODESET V(2)=-15V pro b).

39)

Stabilizátor se Zenerovou diodou – viz též příklad 8. Pomocí příkazu .PRINT zobrazte ve vstupním souboru napětí na R1, výkon na R1 a výkon na diodě.

Poznámka: příkaz .PRINT funguje jen při některé ze základních analýz, např. .DC. Pomocí .PRINT lze přímo zobrazit pouze napětí a proudy, nikoliv tedy výkony. Pomůžeme si např. řízeným zdrojem proudu typu G, zapojeným oběma vývody na referenční uzel, jeho proud vypočteme tak, aby byl roven sledované veličině (třeba výkonu), a následně tento proud zobrazíme příkazem .PRINT.

Stabilizátor se Zenerovou diodou

*

V1 1 0 10V


R1 1 2 250

D1 0 2 D1N750

.DC V1 10 10 1; nastaví se jen jedna hodnota $V1=10V$; krok nesmí být 0

.print DC V(R1) i(GvykonR) i(GvykonD)

GvykonR 0 0 value={V(1,2)^2/250}; výpočet výkonu na R1



GvykonD 0 0 value={-V(2)*i(V1)};výpočet výkonu na D1

.lib

.END

Výsledky: Napětí na R1 je 5.297V, výkon na R1 je 112.2mW, výkon na diodě je 99.65mW.

40)

Získejte tabulku napětí na R1, napětí na D1, výkonů na R1 a výkonů na D1 pro vstupní napětí od 0V do 10V po 1V.

Řešení:

stabil

V1 1 0 10V

R1 1 2 250

D1 0 2 D1N750

.DC V1 0 10 1

.print DC V(R1) V(2) i(GvykonR) i(GvykonD)

GvykonR 0 0 value={V(1,2)^2/250}

GvykonD 0 0 value={-V(2)*i(V1)}

.lib

*

.END

Výsledek:

V1	V(R1)	V(2)	I(GvykonR)	I(GvykonD)
0.000E+00	2.842E-23	-2.842E-23	0.000E+00	0.000E+00
1.000E+00	3.547E-05	1.000E+00	5.032E-12	1.419E-07
2.000E+00	4.595E-04	2.000E+00	8.447E-10	3.675E-06
3.000E+00	5.978E-03	2.994E+00	1.429E-07	7.159E-05
4.000E+00	6.741E-02	3.933E+00	1.818E-05	1.060E-03
5.000E+00	4.589E-01	4.541E+00	8.422E-04	8.335E-03
6.000E+00	1.371E+00	4.629E+00	7.523E-03	2.539E-02
7.000E+00	2.341E+00	4.659E+00	2.191E-02	4.362E-02
8.000E+00	3.322E+00	4.678E+00	4.413E-02	6.216E-02
9.000E+00	4.308E+00	4.692E+00	7.423E-02	8.085E-02
1.000E+01	5.297E+00	4.703E+00	1.122E-01	9.965E-02
1.100E+01	6.288E+00	4.712E+00	1.581E-01	1.185E-01
1.200E+01	7.280E+00	4.720E+00	2.120E-01	1.374E-01
1.300E+01	8.273E+00	4.727E+00	2.738E-01	1.564E-01
1.400E+01	9.267E+00	4.733E+00	3.435E-01	1.754E-01
1.500E+01	1.026E+01	4.739E+00	4.212E-01	1.945E-01

Obvod začíná stabilizovat výstupní napětí při vstupním napětí cca od 5V nahoru. Při vyšších hodnotách vstupního napětí je výkon na R1 větší než výkon na diodě.

Výpis hodnot bývá nepřehledný, mnohdy dáme přednost grafům v PROBE.

Tip: pokud chcete zobrazit čísla na více míst za desetinnou tečkou, umístíme do vstupního souboru příkaz .OPTIONS, např.

.OPTIONS NUMDGT=6

41)

Získejte v PROBE dva obrázky: závislost výkonů na $R1$ a $D1$ na vstupním napětí (obrázek 1) a závislost výstupního napětí na vstupním napětí (obrázek 2).

Řešení:

stabil

V1 1 0 10V

R1 1 2 250

D1 0 2 D1N750

.DC V1 0 15 0.1 ;je zmenšen krok, aby křivky byly dostatečně hladké

GvykonR 0 0 value={V(1,2)^2/250}

GvykonD 0 0 value={-V(2)*i(V1)}

.probe


.lib

.END

Nejprve definujeme spodní obrázek přidáním křivky V(2). Potom přidáme další obrázek (Plot/Add Plot to Window) a definujeme v něm křivky I/(GvykonD) a I/(GvykonR).

Ze spodního obrázku je zřejmé, že obvod nestabilizuje napětí, pokud vstupní napětí poklesne pod cca 5V. Horní obrázek ukazuje, že pro vstupní napětí menší než cca 9.5V je výkon na diodě větší než výkon na rezistoru, pro větší vstupní napětí je tomu naopak.

Horní obrázek sice „ukazuje“ proudy v mA, ve skutečnosti se ale jedná o výkony v mW.

**42)**

Zobrazte výkony na rezistoru a diodě bez mezivýpočtů GvykonR a GvykonD, jen s využitím možností PROBE.

Řešení:

stabil

V1 1 0 10V

R1 1 2 250

D1 0 2 D1N750

.DC V1 0 15 0.1

.probe

.lib

.END

V PROBE zavedeme do obrázku dvě křivky pomocí těchto matematických operací (například):


$(V(1)-V(2)) * I(R1)$

$V(2) * I(R1)$

nebo ještě pohodlněji


W(R1)

W(D1)



43)

Zobrazte ampérovoltovou charakteristiku diody D1N750 pro napětí z intervalu od -4.8V do +0.8V.



Zenerka

Dzener 1 0 D1N750

Vdioda 1 0 1V

.DC Vdioda -4.8 0.8 0.01


.probe

.lib

*

.END

Problém: zdrojem napětí nelze dobře regulovat proud diodou. Lepší by bylo otvírat diodu zdrojem proudu, který budeme rozmítat např. od -100mA do +100mA:



Zenerka

Dzener 1 0 D1N750

Idioda 1 0 100mA

.DC Idioda -100m 100m 1m

.probe

.lib



*

.END

PROBE zobrazuje na vodorovnou osu standardně rozmítanou veličinu. Je třeba „přehodit osy“ a upravit jejich měřítka, vše v okně „Plot/Axis Settings“.


Otázka: proč jsou obě charakteristiky mírně odlišné pro napětí cca od -4.5V do 0V?

(Je to v důsledku toho, že nejprve bylo krokováno napětí a pak proud. Krok byl krokován po 1mA, takže v závěrné oblasti diody bylo použito málo bodů k jejímu vykreslení. Zmenšení kroku by vše napravilo).



44)

Zobrazte sít' výstupních charakteristik tranzistoru Q2N3904. Rozsah napětí kolektor-emitor od 0V do 10V, proud báze od 0 do 100 μ A.



AV char

Qtest 2 1 0 Q2N3904

Vce 2 0 5V

Ib 0 1 100uA


.DC Vce 0V 10V 0.01V Ib 0A 100uA 10uA

.probe

.lib


*

.END

**45)**

Změřte zatěžovací charakteristiku zdroje napětí 10V s napěťovou referencí REF01, tj. závislost výstupního napětí na zatěžovacím proudu.

Tento obvod byl řešen v Př. 25. Níže je použit podobvod REF-01/AD z knihovny profesionální verze OrCadPSpice10.



zdroj 10V

*

Vnap 1 0 15V

R1 2 3 8.77k

R2 3 0 1.23k

Iload 2 0 20mA

Xreference 1 0 3 2 REF-01/AD

.DC Iload 0 20mA 0.1mA


.PROBE

*

.lib

.end

Z výsledků simulace je zřejmé, že daný zdroj spolehlivě pracuje do odběrného proudu cca 17mA.

**46)**


Zjistěte, jak se budou zatěžovací charakteristiky měnit při vstupních napětích od 10V do 20V.

Návod: .DC Iload 0 20m 0.1m Vnap 10 20 2

47)

Zjistěte závislost stejnosměrného kolektorového napětí a kolektorového proudu na teplotě. Teplotu rozmítejte od nuly do 100 stupňů Celsia. Z křivek odečtěte změnu kolektorového napětí na jeden stupeň v mV/°C a procentuální změnu kolektorového proudu na jeden stupeň (nominální hodnotu I_C uvažujte při teplotě 27 stupňů Celsia).

Pozn.: Model tranzistoru je např. v knihovně ebipolar.lib profesionální verze OrCadPSpice 10, nebo si jej získejte např. z Internetu.



Řešení:

zesilDC

*

Vin in 0 AC 1

Cv in base 5u

R2 baterie base 100k

R3 base 0 56k

R1 baterie kolektor 2k

R4 emitor 0 2k

Ce emitor 0 800u

Q1 kolektor base emitor BC107A

Vbat baterie 0 12V

.lib ebipolar.lib

.DC TEMP 0 100 1



.probe Ic(Q1) Vc(Q1)

.end

Z grafů odečteme:

Při 0°C je napětí na kolektoru 8.8092V, při 100°C je napětí 8.5123V. Tedy teplotní součinitel je -2.969mV/°C.

Při 27°C je proud kolektoru 1.6411mA, při 97°C je proud 1.7387°C. Na 1°C to dělá cca nárůst o 0.085%. Pracovní bod je poměrně dobře teplotně stabilizován.

**48)**

Zjistěte závislost kolektorového a emitorového napětí na odporu kolektoru, který rozmítejte od 1kohmu do 10kohmů.

Řešení:

.... Viz příklad 47

R1 baterie kolektor {Rc}

...

.param Rc 2k

.DC param Rc 1k 10k 100

.probe V([kolektor]) V([emitor])

.end

Příklady 49 až 62

Analýza Transient.

Povolování/zakazování výpočtu pracovního bodu.

Zadávání počátečních podmínek u kapacitorů a induktorů.

Příkaz .IC.


Modelování zdrojů signálů.

Příkazy .LOADBIAS, .SAVEBIAS

49)

Zjistěte časový průběh napětí na R, L a C a proudu tekoucího obvodem od okamžiku připojení stejnosměrného zdroje k obvodu. Uvažujte nulové počáteční podmínky. Časový průběh sledujte do času 50 μ s.

Řešení:



příklad49 - RLC transient

Vin 1 0 10V

R 1 2 800

L 2 3 2.53mH

C 3 0 1nF


.TRAN 1u 50u SKIPBP

.PROBE V(R) V(L) V(C) I(R)

.end

SKIPBP znamená zákaz výpočtu pracovního bodu před vlastní analýzou Transient. Zkuste vyblokovat – SPICE najde přímo stejnosměrný ustálený stav. Zkuste zaměnit za ekvivalent – UIC (Use Initial Conditions).

Všimněte si nepřesně vykreslených časových průběhů v oblastech „špiček“ průběhů.

**50)**

Pokuste se lépe vykreslit detaily křivek pomocí parametru „step ceiling“.

Řešení:

Příklad50 - RLC transient

Vin 1 0 10V

R 1 2 800

L 2 3 2.53mH


C 3 0 1nF

.TRAN 1u 50u 0 0.1u SKIPBP

.PROBE V(R) V(L) V(C) I(R)

.end

Dvě vložená čísla značí: čas, do něhož je potlačen výstup výsledků simulace, a „step ceiling“, t.j. maximální povolený krok časové analýzy. Bez uvedení těchto čísel platí implicitní hodnoty 0, Tstop/50, neboli 0, 1 μ s.

**51)**

Pokuste se lépe vykreslit detaily křivek zpřesněním analýzy, konkrétně snížením chybového kritéria RELTOL (standardně nastaveno na 0.001).


Řešení: Analýzu inicializujeme bez modifikace „step ceiling“, viz př. 49, ale přidáme příkaz

.OPTIONS RELTOL 1e-6

52)

Po ustálení přechodných dějů v obvodu z příkladu 49 bude napětí na kapacitou 10V a obvodem nepoteče proud. Určete přechodný děj, který vznikne, když vstupní napětí náhle klesne na nulu.

Pozn.: Místo zdroje s nulovým napětím je možno použít zkrat.



Řešení:

příklad52 - RLC transient

Vin 1 0 0V

R 1 2 800


L 2 3 2.53mH

C 3 0 1nF IC=10V

.TRAN 1u 50u 0 0.1u SKIPBP

.PROBE V(R) V(L) V(C) I(R)

.end

**53)**

Příklad 52 vyřešte s využitím příkazu .IC.

Řešení:

příklad53 - RLC transient

Vin 1 0 0V

R 1 2 800

L 2 3 2.53mH

C 3 0 1nF

.IC V(3)=10V

.TRAN 1u 50u 0 0.1u SKIPBP

.PROBE V(R) V(L) V(C) I(R)

.end

Poznámky:

Ověřte si, že daný příkaz .IC V(3)=10V vede na stejné výsledky jako například příkazy

.IC V(3)=10V I(L)=0A

.IC V(3)=10V V(2)=0V

.IC V(3)=10V V(2)=10V

Pokuste se vysvětlit „nesrovnalost“ v posledních dvou příkazech.


Pokuste se vysvětlit, proč po uvedení příkazu

.IC V(3)=10V I(L)=0A

můžeme odstranit slovo SKIPBP a výsledek analýzy se nezmění.

54)

Při odporu 10Ω teče obvodem v ustáleném stavu proud 1A. Nakreslete přechodný děj, který nastane, když se náhle odpor zvětší na $1k\Omega$ (simulujeme rozpojení spínače). Sledujte napětí na L a proud obvodem.



Řešení:

příklad54 - RL transient

Vin 1 0 0V

R 1 2 1k


L 2 0 10mH IC=1A

.TRAN 1u 50u SKIPBP

.PROBE V(R) V(L) I(R)


.end

Na cívce se objeví záporná napěťová špička 1kV. Vysvětlete.



55)

Zobrazte přechodný děj v tranzistorovém zesilovači po jeho připojení k napájecí baterii. Před připojením baterie je zesilovač v nulovém energetickém stavu. Vstupní signál je nulový.



Řešení:

příklad55 - zesilovac

Vin in 0 0V

Cv in b 10u

Rb1 bat b 60k

Rb2 b 0 44k

Rc bat c 3.3k

Re e 0 3.3k

Ce e 0 100u

Vbat bat 0 10V


Q1 c b e Q2N2222

.TRAN 1u 2 SKIPBP

.PROBE

.lib

.end



Závěr: Trvá skoro dvě sekundy, než je stejnosměrný pracovní bod „připraven“.

Ověřte si, že po blokování slova SKIPBP začne analýza TRANSIENT přímo z pracovního bodu, vypočteného iterační procedurou (je potlačen přechodný děj).

56)

Zobrazte přechodný děj v tranzistorovém zesilovači po jeho připojení k napájecí baterii. Před připojením baterie je zesilovač v nulovém energetickém stavu. Současně s baterií se připojí na vstup signál, který je sinusový o amplitudě 50mV a kmitočtu 10Hz.

Řešení:

příklad56 - zesilovac

Vin in 0 SIN 0 50mV 10Hz

Cv in b 10u

Rb1 bat b 60k

Rb2 b 0 44k

Rc bat c 3.3k

Re e 0 3.3k

Ce e 0 100u

Vbat bat 0 10V

Q1 c b e Q2N2222



.TRAN 1u 2 0 1m SKIPBP; zjemnění kroku

.PROBE

.lib

.end

Z časového průběhu odečteme rozkmit napětí na kolektoru cca 1.898V špička-špička, což je amplituda 0.949V. To představuje střídavé zesílení na kmitočtu 10Hz cca $0.949V/50mV=18.98$.

**57)**

Potlačte „pomalý“ přechodný děj náběhu zesilovače do stejnosměrného pracovního bodu povolením jeho výpočtu před analýzou Transient.


Řešení: Upravíme pouze příkaz .TRAN:

.TRAN 1u 2 0 1m

58)

Na vstup komparátoru s hysterezí připojte zdroj symetrických trojúhelníkových kmitů, napětí od -10V do +10V, kmitočet 1Hz. Prozkoumejte průběh výstupního napětí komparátoru a zobrazte hysterezní charakteristiku komparátoru $U_2=f(U_1)$.

Řešení:



příklad58 - komparator

Vin 1 0 PULSE -10 10 0 0.5 0.5 1p 1 ;misto 1p ma byt teoreticky nula (sirka impulsu); pak ale PSPICE misto nuly dosadi implicitni hodnotu Tstop.

Xkomp 3 1 5 4 2 LF411

Vplus 5 0 15V

```

Vminus 0 4 15V
R1 2 3 1k
R2 3 0 1k
.TRAN 1m 2 ;vykresli se 2 opakovací periody
                ;pocitani pracovniho bodu povoleno

.PROBE
.lib
.end

```

V PROBE zobrazíme V(1) a V(2). Pak založíme další obrázek, do něhož umístíme hysterezní charakteristiku. Protože na vodorovnou osu nyní vyneseme V(1), nikoliv čas, je třeba zadat Plot/Unsynchronize X Axis


Náměty na další práci:

Vysvětlete, jak PSPICE dospěl k prvnímu bodu převodní charakteristiky. Jaký vliv na analýzu bude mít klíčové slovo SKIPBP v příkaze .TRAN?

Zkuste „zrychlit“ rozmítání vstupního signálu, např.

```
Vin 1 0 PULSE -10 10 0 5u 5u 1p 10u
```

```
.TRAN 1n 20u
```



Přesvědčte se o tom, že se nyní výrazně uplatní mezní rychlost přeběhu OZ a převodní charakteristika již bude silně deformovaná (již to nebude stejnosměrná charakteristika, kterou jsme chtěli získat).

Pokuste se realizovat rozmítání komparátoru zdrojem typu „PWL“.

59)

V příkladu 58 odstraňte zdroj vstupního signálu a komparátor doplňte zpětnovazebními součástkami R a C podle obr. Jde o generátor obdélníkových a pilovitých kmitů o kmitočtu

$$f \approx 0.455 / (RC) \approx 98.6 \text{ Hz.}$$

Prostudujte možnosti rozběhu generátoru do provozního režimu.

Řešení:

příklad59 - funkční generator

```
Xkomp 3 1 5 4 2 LF411
```

```
Vplus 5 0 15V
```

```
Vminus 0 4 15V
```

```
R1 2 3 1k
```

```
R2 3 0 1k
```

```
R 2 1 100k
```


```
C 1 0 47nF
```

```
.TRAN 1n 60m 0 100u SKIPBP
```

```
.PROBE
```

```
.lib
```

```
.end
```



Změřte opakovací periodu kmitů a z ní odvoďte kmitočet. Možné správné výsledky: 10.336ms, 96.75Hz. ZV důsledku numerických chyb mírně závisí na pořadí měřené periody. Pokuste se periodu změnit i pomocí měřicích funkcí Period, příp. Period_XRange.

Náměty na samostatnou práci:

Ověřte si, jaké problémy má PSpice s nasazováním kmitů, pokud mu ponecháme přednastavenou hodnotu max. časového kroku (Step ceiling), tj.

.TRAN 1n 60m SKIPBP

Obvod nyní nabíhá do ustálených kmitů podstatně déle (je třeba zvětšit simulační čas). Simulátoru můžeme „pomoci“ například nastavením počátečního napětí na kapacitě na 100mV.

Pokuste se rozběhnout generátor při povolení výpočtu pracovního bodu:

.TRAN 1n 60m 0 100u

Postupně zkoušejte tyto nastavené počáteční podmínky a přemýšlejte o získaných výsledcích:

.IC V(2)=15V

.IC V(2)=15V V(1)=0V

.IC V(2)=-15V

.IC V(2)=-15V V(1)=0V

60)

Prostudujte rozběh oscilátoru s tranzistorem MOSFET z nulových počátečních podmínek.

Pozn.: Prostudujte syntaxi zadávání tranzistoru MOSFET.

Řešení:

příklad60 - RC oscilátor s MOSFETem

M1 D G 0 0 M2N6661

Vbat bat 0 10V

R1 D 1 10k

R2 1 2 100k

R3 2 G 1meg

C1 1 0 28.4nF

C2 2 0 2.84nF

C3 G 0 284pF

R4 bat D 100




.TRAN 1n 20m 0 20u SKIPBP

.PROBE

.lib

.end

Ověřte, že kmitočet generovaných kmitů je asi 971Hz.



61)

Pomocí příkazů `.SAVEBIAS` a `.LOADBIAS` dosáhněte přímo simulace ustálených kmitů oscilátoru bez mezivýpočtů přechodného děje.

Řešení:

1. Vytvoření souboru s výsledky simulace na konci analyzačního běhu analýzy Transient:

příklad61 - RC oscilator s MOSFETem

.....

.SAVEBIAS 60.ope TRAN TIME=20ms


.TRAN 1n 20m 0 20u SKIPBP

.....

2. Editace vzniklého souboru záměnou `.NODESET` za `.IC`

3. Spuštění simulace s modifikovaným příkazem:

.LOADBIAS 60.ope



Poznámka: příkaz `.IC` v externím souboru je možno překopírovat do našeho vstupního souboru. Pak již k simulaci nebudeme externí soubor potřebovat (a tím ani příkaz `.LOADBIAS`):

příklad61 - RC oscilator s MOSFETem

M1 D G 0 0 M2N6661

Vbat bat 0 10V

R1 D 1 10k

R2 1 2 100k

R3 2 G 1meg

C1 1 0 28.4nF

C2 2 0 2.84nF

C3 G 0 284pF

R4 bat D 100

`.IC`

+ V(1) = 2.0676387445

+ V(2) = 3.1072366239

+ V(D) = 1.0196013703

+ V(G) = 3.4454667233

+ V(bat) = 10.0000000000

.TRAN 1n 20m 0 20u SKIPBP

.PROBE

`.lib`

`.end`

62)

Simulujte odezvu zesilovače na skokovou změnu vstupního napětí z 0V na 1V pro odpory $R1=R2=(1000, 400, 300, 200)$ ohmů, nejprve bez a pak s uvažováním parazitní kapacity 10pF.

Uvažte, že AD8001 je velmi rychlý OZ typu CFA s dobou ustálení jednotek až desítek ns a že dynamické vlastnosti zesilovače budou záviset nejen na poměru $R1/R2$, ale i na absolutních velikostech odporů. Čím menší odpory, tím větší náchylnost k nestabilitám.

Řešení:

Zesilovác s AD8001

Vin in 0 PWL (0,0) (1n,0)(1.1n 1) ;jednotkový skok z 0V na 1V v čase 1ns, nástupná hrana 0.1ns

Vplus Vp 0 5V

Vminus Vn 0 -5V

R1 out inv {Rx}

R2 inv 0 {Rx}

*Cpar inv 0 10p

X in inv Vp Vn out AD8001a/AD

.param Rx 10

.step param Rx list 1k 400 300 200




.TRAN 1p 6n 0 100p ;časová analýza do 6ns s max. časovým krokem 100ps

.probe

.lib

.end

Při $R1=R2=1k$ ohm se výstupní napětí ustálí bez překmitu za cca 3.5ns. Při $R1=R2=200$ ohmů je zesilovač potenciálně nestabilní, protože např. při parazitní kapacitě 10pF dojde k netlumeným oscilacím. Pozn.: Při simulaci vlivu parazitní kapacity zvolte delší simulační čas, např. 20ns.



Příklady 63 až 74

Analýza AC.

Příkaz .AC.

Příkaz .PLOT v analýze AC

Vykreslování kmitočtových charakteristik amplitudových a fázových, klasických, Bodeho, komplexních.

63)

Zjistěte střídavá napětí a proudy u všech součástek Wienova článku, je-li kmitočet střídavého vstupního napětí 1kHz, amplituda 10 a počáteční fáze 0. Zjistěte amplitudy a počáteční fáze těchto veličin a jejich reálné a imaginární složky. Nakreslete fázorové diagramy.

Řešení:

Wienův článek na 1kHz

Vin 1 0 Ac 1

R1 1 2 1k

C1 2 3 159nF

R2 3 0 1k

C2 3 0 159nF

.AC LIN 1 1k 1k ;AC analýza proběhne na jediném kmitočtu 1kHz

.print AC V(R1) VP(R1) V(C1) VP(C1) V(R2) VP(R2) ; výstup modulu a fáze jednotlivých napětí

.print AC VR(R1) VI(R1) VR(C1) VI(C1) VR(R2) VI(R2) ; výstup reálných a imag. složek napětí

.print AC I(R1) IP(R1) I(R2) IP(R2) I(C2) IP(C2) ; výstup modulu a fáze jednotlivých proudů

.print AC IR(R1) II(R1) IR(R2) II(R2) IR(C2) II(C2) ; výstup reálných a imag. složek proudů

.end

Výsledky z výstupního souboru (zkráceno):

FREQ	V(R1)	VP(R1)	V(C1)	VP(C1)	V(R2)	VP(R2)
1.000E+03	4.712E-01	4.501E+01	4.716E-01	-4.499E+01	3.333E-01	3.720E-02

FREQ	VR(R1)	VI(R1)	VR(C1)	VI(C1)	VR(R2)	VI(R2)
1.000E+03	3.331E-01	3.332E-01	3.336E-01	-3.334E-01	3.333E-01	2.164E-04

FREQ	I(R1)	IP(R1)	I(R2)	IP(R2)	I(C2)	IP(C2)
1.000E+03	4.712E-04	4.501E+01	3.333E-04	3.720E-02	3.330E-04	9.004E+01


FREQ	IR(R1)	II(R1)	IR(R2)	II(R2)	IR(C2)	II(C2)
1.000E+03	3.331E-04	3.332E-04	3.333E-04	2.164E-07	-2.162E-07	3.330E-04

64)

Do výstupního souboru vytiskněte tabulku závislosti amplitudy a počáteční fáze výstupního napětí V(3) na kmitočtu v kmitočtových bodech 100Hz, 1kHz a 10kHz.

Výsledky:

FREQ	V(3)	VP(3)
1.000E+02	9.658E-02	7.316E+01
1.000E+03	3.333E-01	3.720E-02
1.000E+04	9.676E-02	-7.313E+01



Řešení:

Wienův článek na 1kHz

Vin 1 0 Ac 1

R1 1 2 1k

C1 2 3 159nF

R2 3 0 1k

C2 3 0 159nF

.AC DEC 1 100 10k

.print AC V(3) VP(3)

.end

Závěr: Zdá se, že v okolí kmitočtu 1kHz má Wienův článek největší přenos $1/3$ a nulový fázový posuv mezi vstupním a výstupním napětím. Tento kmitočet je dán vzorcem $1/(2\pi RC)$.

65)

Zobrazte amplitudovou a fázovou kmitočtovou charakteristiku Wienova článku ve frekvenčním rozsahu od 10Hz do 100kHz (logaritmická frekvenční osa, logaritmické rozmítání kmitočtu).

Řešení:

Wienův článek, kmitočtové charakteristiky

Vin 1 0 Ac 1

R1 1 2 1k

C1 2 3 159nF


R2 3 0 1k

C2 3 0 159nF

.AC DEC 10 10 100k

.probe V(3)

.end



Pozn.: v PROBE je amplitudová charakteristika zobrazena jednak jako klasický přenos napětí a jednak v decibelech. Každá křivka má svou vlastní osu Y (Plot/Add Y Axis).

66)



Zobrazte komplexní kmitočtovou charakteristiku Wienova článku z příkladu 65.

Řešení: V PROBE je třeba nastavit na vodorovnou osu zobrazování reálné části přenosu a na svislou osu zobrazování imaginární části přenosu.

Pro vykreslení hladké křivky je vhodné zvýšit počet kroků na dekádu na cca 100.

67)

Zjistěte stejnosměrná a střídavá napětí ve všech uzlech zesilovače, je-li kmitočet vstupního signálu 10kHz.



Řešení:

Tran. zesilovac

Vin in 0 AC 20mV

Cv in baze 330nF

Rb1 bat baze 100k

Rb2 baze 0 56k

Rc bat kol 2k

Re emi 0 2k

Ce emi 0 500uF

Vbat bat 0 12V

Q kol baze emi BC107A

.AC LIN 1 10k 10k

.print AC V([in]) V([baze]) V([kol]) V([emi]) V([bat])

.lib

.end

Výsledky:

Ss pracovní bod:

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(in)	0.0000	(bat)	12.0000	(emi)	3.2979	(kol)	8.7207	(baze)	3.9738

Střídavý signál – amplitudy:

FREQ	V(in)	V(baze)	V(kol)	V(emi)	V(bat)
1.000E+04	2.000E-02	1.997E-02	2.464E+00	3.942E-05	1.000E-30

Střídavý signál na kolektoru nebude ořezán, protože ss napětí na kolektoru 8.7207V je „vzdáleno“ od napětí baterie asi 3.28V a amplituda na kolektoru je 2.464V.

Střídavé zesílení je asi $2.464/0.02=123.2$.

Přesvědčte se, že fázový posuv napětí na kolektoru oproti vstupnímu napětí je asi -179 stupňů, tj. prakticky 180 stupňů (invertováno).

Návod: .print VP(v[kol]).

68)

Analýzujte amplitudovou a fázovou kmitočtovou charakteristiku zesilovače z příkladu 67 v kmitočtovém rozsahu 100Hz až 100MHz.

Řešení:

Tran. zesilovac

Vin in 0 AC 1V

Cv in baze 330nF

Rb1 bat baze 100k


Rb2 baze 0 56k

Rc bat kol 2k

Re emi 0 2k

Ce emi 0 500uF

Vbat bat 0 12V



```

Q kol baze emi BC107A
.AC DEC 10 100 100meg
.probe V([kol])
.lib
.end

```

Ověřte si, že na kmitočtu 10kHz je zesílení asi 41.813 dB, tj. zesílení 123.2, a fázový posuv -178.956 stupňů. Porovnejte s výsledky z příkladu 67.

69)


Zobrazte amplitudové kmitočtové charakteristiky zesilovače s výstupem na kolektoru a na emitoru pro blokovací emitorovou kapacitu 1pF, 1uF a 500uF.

Řešení:

```

Tran. zesilovac
Vin in 0 AC 1V
Cv in baze 330nF
Rb1 bat baze 100k
Rb2 baze 0 56k
Rc bat kol 2k
Re emi 0 2k
Ce emi 0 {Cx}
Vbat bat 0 12V
Q kol baze emi BC107A
.param Cx 500u
.step param Cx list 1p 1u 500u
.AC DEC 10 100 100meg
.probe V([kol]) V([emi])
.lib
.end

```



Při zanedbatelné kapacitě C_e se zesilovač chová z hlediska emitoru jako sledovač (zesílení +1) a z hlediska kolektoru jako invertor (zesílení -1), a to až do desítek MHz. Při růstu C_e roste zesílení na kolektoru, neboť je blokována záporná zpětná vazba přes R_e .

70)


Zobrazte amplitudovou a fázovou kmitočtovou charakteristiku operačního zesilovače typu 741.

Řešení:

```

Operacni zesilovac 741
Vin in 0 AC 1V
Vplus Vp 0 15V
Vminus Vn 0 -15V
X in 0 Vp Vn out uA741
.AC DEC 10 0.1 100meg
.probe
.lib
.end

```



Ověřte tyto číselné údaje:

Zesílení na nízkých kmitočtech je asi 106dB, tj. 200000.

Kmitočet třidecibellového poklesu zesílení je asi 5Hz. Fázový posuv výstupního napětí oproti vstupnímu je na tomto kmitočtu asi 45 stupňů.

Tranzitní kmitočet, tj. kmitočet, při kterém poklesne zesílení na 0 dB, je asi 900kHz.

Mezi kmitočtem třidecibellového poklesu a kmitočtem tranzitním klesá zesílení se strmostí 20 db/dekádu neboli 6 db/oktávu.

Za tranzitním kmitočtem se nachází druhý lomový kmitočet, který způsobuje dvojnásobně rychlý pokles zesílení 40 db na dekádu neboli 12 db na oktávu.

Poznámka: strmost poklesu zesílení lze měřit v PROBE například měřicí funkcí Swing_XRange.

Příklady:

```
Swing_XRange(db(V(out)),10meg,100meg)
```

(měření strmosti 40 db/dekádu za kmitočtem 2. lomu)

```
Swing_XRange(db(V(out)),10meg,20meg)
```

(měření strmosti 12 db/oktávu za kmitočtem 2. lomu)

```
Swing_XRange(db(V(out)),1k,10k)
```

(měření strmosti 20 db/dekádu před kmitočtem 2. lomu).

71)

Nakreslete amplitudové kmitočtové charakteristiky zesilovače pro $R_2=1k$, $R_1=1m$, $1k$, $10k$, $100k$ ohmů.

Řešení:

Operacni zesilovac 741

Vin in 0 AC 1V

Vplus Vp 0 15V

Vminus Vn 0 -15V

R1 out inv {Rx}

R2 inv 0 1k

X in inv Vp Vn out ua741

.param Rx 10k

.step param Rx list 1m 1k 10k 100k

.AC DEC 10 100 100meg




.probe

.lib

.end

Ověřte, že stejnosměrné zesílení je dáno vzorcem

$A_0=1+R_1/R_2$ a kmitočet třidecibellového poklesu je GBW/A_0 , kde GBW je asi 1MHz.



72)

Nakreslete amplitudové kmitočtové charakteristiky zesilovače s CFA operačním zesilovačem AD8001 pro odpory $R_1 = 1k$, $R_2 = 100$, $1k$, $1G$.

Řešení:

Operacni zesilovac AD8001

Vin in 0 AC 1V

Vplus Vp 0 5V

Vminus Vn 0 -5V

R1 out inv 1k

R2 inv 0 {Rx}

X in inv Vp Vn out **AD8001A/AD**

.param Rx 10k

.step param Rx list **100 1k 1G**

.AC DEC 10 **1k 1000meg**

.probe

.lib

.end



Poznámky:

Tento operační zesilovač je „rychlejší“ než uA741, má daleko vyšší tranzitní kmitočet.

Způsob řízení zesílení odpory je jiný než u napěťového OZ. Řídíme-li zesílení odporem R_2 , pak je zachována šířka pásma zesilovače.

Ověřte, že stejnosměrné zesílení je dáno vzorcem

$$A_0 = 1 + R_1/R_2$$

**73)**

Zapojte operační zesilovač AD8001 jako neinvertující zesilovač se zesílením 2. Změřte jeho amplitudovou kmitočtovou charakteristiku v kmitočtovém rozsahu od 10 MHz do 400 MHz pro odpory $R_1 = R_2 = (649, 698, 750)$ ohmů.

Řešení:

Operacni zesilovac AD8001

Vin in 0 AC 1V

Vplus Vp 0 5V

Vminus Vn 0 -5V

R1 out inv {Rx}

R2 inv 0 {Rx}

X in inv Vp Vn out **AD8001a/AD**

.param Rx 10k



.step param Rx list 649 698 750

.AC OCT 10 10meg 400meg

.probe

.lib

.end




Poznámka: U operačních zesilovačů CFA záleží nejen na poměrech odporů R_1 a R_2 , ale i na jejich absolutních velikostech. Je třeba optimalizovat zejména zpětnovazební odpor R_1 . Čím menší R_1 , tím větší šířka pásma, ale tím větší překmity v kmitočtové charakteristice a náchylnost k nestabilitě. Hodnoty odporů jsou převzaty z katalogového listu AD8001. Zkontrolujte s výsledky simulace.

74)

Analýzujte amplitudovou a fázovou kmitočtovou charakteristiku otevřené smyčky zesilovače pro $R_1=R_2=(1000, 400, 300, 200)$ ohmů a pro parazitní kapacitu $C_{par}=10\text{pF}$.

Z charakteristik odečtěte fázovou bezpečnost pro všechny 4 případy a rozhodněte o potenciální stabilitě zesilovače. Porovnejte s výsledky z příkladu 62.

Poznámka: Fázová bezpečnost je doplněk fázového posunu mezi výstupním a vstupním napětím zesilovače do 180 stupňů na kmitočtu, při němž je zesílení 1, tj. 0 db. Čím je fázová bezpečnost větší, tím lépe pro stabilitu obvodu.



Řešení:

Operacni zesilovac AD8001

Vin in 0 AC 1V

Vplus Vp 0 5V

Vminus Vn 0 -5V

R1 out inv {Rx}

R2 inv 0 {Rx}

Cpar inv 0 10p

X in inv Vp Vn out AD8001a/AD

.param Rx 200


.step param Rx list 1000 400 300 200

.AC OCT 100 100k 10G

.probe

.lib

.end



Evaluate	Measurement	1	2	3	4
☒	PhaseMargin(db(v(inv)/v(in, inv)), P(v(i...	44.06162	16.36399	8.08168	-3.50735

Fázovou bezpečnost změříme buď pomocí kurzorů nebo lépe pomocí měřicí funkce PhaseMargin. Pro odpory 200 ohmů je již fázová bezpečnost záporná, což znamená nestabilitu.

Poznámka: Přenos tzv. otevřené smyčky (Open Loop Gain) je přenos od vstupních svorek OZ k výstupu děliče R_1 - R_2 , tj. $V(\text{inv})/V(\text{in}, \text{inv})$.

Ověřte si, že při parazitní kapacitě 7pF a méně je již fázová bezpečnost vždy kladná pro všechny uvažované odpory.

Příklady 75 až 80

Analýza TF – Transfer Function (přenosová funkce).

Příkaz .TF.

Hledání parametrů Théveninova modelu obvodu pomocí příkazu .TF.

Hledání malosignálových charakteristik nelineárních obvodů pomocí příkazu .TF.

Analýza .SENS – Sensitivity Analysis (citlivostní analýza).

75)

Rezistory R1, R2 a R3 tvoří zeslabovač TV signálu. Ověřte, že vstupní odpor zeslabovače bude 50 ohmů, jestliže výstupní svorky budou zatíženy opět 50 ohmy (impedanční přizpůsobení koaxiálními kabelemi), a to za předpokladu, že odpory v zeslabovači jsou navrženy podle vzorců

$$R_1 = R_2 = R = \frac{50}{\sqrt{1+2a}}, \quad R_3 = aR, \quad a \text{ je libovolné číslo větší než } 0 \text{ a závisí na něm zeslabení signálu.}$$


Dále ověřte, že výstupní odpor zeslabovače je rovněž 50 ohmů.

Ověřte pro $a = 4, 12, 24$. Pro tyto hodnoty stanovte zeslabení signálu.

Pozn.: Vstupní odpor zeslabovače 50 ohmů je to samé jako odpor 100 ohmů naměřený na svorkách zdroje V_{in} . Výstupní odpor zeslabovače 50 ohmů je to samé jako odpor 25 ohmů naměřený na svorkách zátěže R_{out} .

Napětí V_{in} se 2x zeslabí na svorkách 2-0, takže zeslabení signálu je ve skutečnosti 2x větší než zeslabení obvodem R1-R2-R3.

Řešení:



Zeslabovac

Vin 1 0 1V

Rin 1 2 50

R1 2 3 {R}

R2 3 4 {R}

R3 3 0 {a*R}

Rout 4 0 50

.param a 12

.param R {50/sqrt(1+2*a)}

.step param a list 4 12 24

.TF V(4) Vin

.end

Výsledky – uvedeno v zhuštěné formě:

PARAM A=4

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(1)	1.0000	(2)	.5000	(3)	.3333	(4)	.2500

V(4)/Vin = 2.500E-01

INPUT RESISTANCE AT Vin = 1.000E+02

OUTPUT RESISTANCE AT V(4) = 2.500E+01

PARAM A=12

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(1)	1.0000	(2)	.5000	(3)	.4000	(4)	.3333

V(4)/Vin = 3.333E-01

INPUT RESISTANCE AT Vin = 1.000E+02

OUTPUT RESISTANCE AT V(4) = 2.500E+01

PARAM A=24

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(1)	1.0000	(2)	.5000	(3)	.4286	(4)	.3750

V(4)/Vin = 3.750E-01


INPUT RESISTANCE AT Vin = 1.000E+02

OUTPUT RESISTANCE AT V(4) = 2.500E+01

76)

Pomocí příkazu `.TF` nalezněte Théveninův model obvodu vzhledem k jeho svorkám A-GND, tj. nalezněte velikost napětí naprázdno na svorce A a výstupní odpor.

Obvod představuje náhradní lineární model tranzistorového zesilovače v pásmu středních kmitočtů.



Řešení:

Thevenin

Vin 1 0 1V

Rbe 1 3 2k

Re 3 0 100

Rb 1 2 100k

Rc 2 0 1k

G 2 3 1 3 0.1

.TF V(2) Vin

.end

Výsledky:

V(2)/Vin = -8.950E+00

INPUT RESISTANCE AT Vin = 6.908E+03

OUTPUT RESISTANCE AT V(2) = 9.901E+02

77)

Určete střídavé zesílení v pásmu středních kmitočtů, je-li výstup zesilovače na emitoru.

Vazební kapacitor znemožní výpočet zesílení příkazem `.TF`, protože obvodové veličiny se počítají jako stejnosměrná řešení. Klasické použití `.TF` vede k nesprávným výsledkům:

Transfer function

Vin in 0 ; napětí je 0V, ve skutečnosti na něm nezáleží

Cv in baze 330nF

Rb1 bat baze 100k

Rb2 baze 0 56k

Rc bat kol 2k

Re emi 0 2k


Q kol baze emi BC107A

Vbat bat 0 12V

.TF V([emi]) Vin

.lib

.end



Výsledky:

Ss pracovní bod

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(in)	1.0000	(bat)	12.0000	(emi)	3.2979	(kol)	8.7207	(baze)	3.9738

V(emi)/Vin = 0.000E+00

INPUT RESISTANCE AT Vin = 1.000E+20

OUTPUT RESISTANCE AT V(emi) = 1.808E+02

Přenos napětí je nulový, protože signál „neprošel“ přes Cv. Z téhož důvodu je vstupní odpor prakticky nekonečný. Výstupní odpor je určen při vstupu naprázdno, což neodpovídá režimu zesilovače v pásmu středních kmitočtů.

Klidové napětí na bázi je 3.9738V. V ss ustáleném stavu se tedy C_v chová jako baterie. Po náhradě C_v baterií dostaneme správné výsledky, odpovídající malosignálovým parametrům zesilovače v pásmu středních kmitočtů:

Transfer function

V_{in} in 0 ; nyní bude výsledek správný jen při napětí 0V

Vpom baze in 3.9738V

Rb1 bat baze 100k

Rb2 baze 0 56k

Rc bat kol 2k

Re emi 0 2k


Q kol baze emi BC107A

Vbat bat 0 12V

.TF V([emi]) Vin

.lib

.end



Výsledky:


$$V(\text{emi})/V_{in} = 9.918\text{E-}01$$

$$\text{INPUT RESISTANCE AT } V_{in} = 3.284\text{E+}04$$

$$\text{OUTPUT RESISTANCE AT } V(\text{emi}) = 1.600\text{E+}01$$

78)

Vypočítejte citlivost výstupního napětí stabilizátoru na vstupní napětí a odpory R a R_z .



Řešení:

DC SENSITIVITIES OF OUTPUT V(2)

Stabilizator	ELEMENT NAME	ELEMENT VALUE	ELEMENT SENSITIVITY (VOLTS/UNIT)	NORMALIZED SENSITIVITY (VOLTS/PERCENT)
Vin 1 0 10V				
R 1 2 250				
D1 0 2 D1N750				
Rz 2 0 1k				
.SENS V(2)				
.lib	D1			
.end	SERIES RESISTANCE			
	RS	2.500E-01	1.629E-02	4.072E-05
	INTRINSIC PARAMETERS			
	IS	8.805E-16	-2.907E+00	-2.559E-17
	N	1.000E+00	0.000E+00	0.000E+00

Výstupní napětí se zvýší o 1.261mV, když se vstupní napětí zvýší o 1% nad 10V, tedy o 100mV. Činitel stabilizace je tedy $100/1.261=79.3$.

79)

Vypočítejte citlivosti napětí na kolektoru a proudu kolektorem na odpory R_b , R_c a parametr BF tranzistoru.


Poznámka: Abychom mohli počítat citlivost proudu, je třeba do série s R_c vložit pomocný zdroj napětí o nulovém napětí a počítat citlivost proudu tímto zdrojem.

Řešení:

Zesilovac

Vin vstup 0

Cv vstup baze 5uF



```

Rb bat baze 757k
Rc bat pom 2k
Vpom pom vystup
Q vystup baze 0 BC107A
Vbat bat 0 12V
.SENS V([vystup]) I(Vpom)
.lib
.end

```


Výsledky:

Relativní citlivost kolektorového napětí na R_b , R_c a B_F je 60.74mV, -54mV, -29.19mV.

Relativní citlivost kolektorového proudu na R_b , R_c a B_F je -30.37uA, -1.251uA, 14.59uA.

80)

Zapojení zesilovače z příkladu 79 je změněno tak, že v obvodu působí záporná zpětná vazba. Odpor R_b je navržen tak, aby nedošlo k podstatné změně ss pracovního bodu (můžete si ověřit). Vypočtěte citlivosti stejně jako v příkladu 79 a porovnejte výsledky. V důsledku stabilizační zpětné vazby by měly být citlivosti menší.



Řešení:

```

Zesilovac
Vin vstup 0
Cv vstup baze 5uF
Rb pom baze 380k
Rc bat pom 2k
Vpom pom vystup
Q vystup baze 0 BC107A
Vbat bat 0 12V
.SENS V([vystup]) I(Vpom)
.lib
.end

```

Výsledky:

Relativní citlivost kolektorového napětí na R_b , R_c a B_F je 29.26mV, -26.09mV, -13.99mV.

Relativní citlivost kolektorového proudu na R_b , R_c a B_F je -14.56uA, -15.16uA, 7.03uA.

Citlivosti jsou zhruba 2x nižší s výjimkou citlivosti I_c na R_c , která vzrostla. Pokuste se o vysvětlení.

Jaké jsou praktické závěry z této analýzy o teplotní stabilitě obvodu?

Příklady 81 až 91

Analýza .FOUR – Fourier Analysis (Fourierova analýza – výpočet spekter signálů).

Zjišťování činitele harmonického zkreslení (THD) signálů.

Studium intermodulačního zkreslení signálů.

Analýza .NOISE – Noise Analysis (Šumová analýza).

Úvod do vyhodnocovací analýzy

Tip pro použití PROBE k spektrální analýze:

Chceme-li zobrazit spektrum signálu v samostatném okně, pak nepoužijeme ikonu , ale zvolíme „Plot/Add Plot To Windows“ (nebo ikonu ) . Poté navolíme „Fourier Transform“ v „Plot Window Templates“ a za argument vybereme příslušný signál.

81)

Pomocí zdroje typu E vygenerujte harmonické napětí typu „sinus“ o amplitudě 3V a kmitočtu 1kHz. Zobrazte pět opakovacích period. Vypočítejte amplitudy prvních 9 harmonických složek. Spektrum amplitud zobrazte v PROBE.

Řešení:

Zdroj sinus

.param pi=3.14159





Esinus 1 0 value={3*sin(2*pi*1k*time)}

.tran 1u 5m

.FOUR 1k V(1)

.probe

.end



Přesvědčte se o tom, že ve výstupním souboru je 1. harmonická vypočtena přesně 3V a další harmonické že jsou zanedbatelné (mají být nulové, hodnoty jsou dány numerickými nepřesnostmi). Počáteční fáze 1. harmonické vyšla 0 (standardně SPICE chápe sinusovku jako signál s nulovou poč. fází). Stejnoseměrná složka i THD by teoreticky měly být nulové.

V PROBE kliknutím na ikonu  zobrazíme spektrální čáry signálu. Je však třeba upravit měřítko na ose x, např. od 0 do 5kHz. PROBE proloží vypočtené body lomenou čarou.

Pozn.: V PROBE se počítá spektrum odděleně od SPICE. Za 1 periodu je považován celý zobrazený úsek signálu, tj. 5mS. Z toho vychází opakovací kmitočet 200Hz. PROBE tedy považuje 200Hz za kmitočet základní harmonické. Na tomto kmitočtu ale spočítá prakticky nulovou spektrální složku. Nenulová je až na „pravém“ kmitočtu 1kHz..

Mnohdy je lepší přepnout osu Y na logaritmickou. Zvýrazní se složky, které nejsou v lin. měřítku viditelné.

Námět na další experimenty:

Co se stane, když nepřesně stanovíme opakovací kmitočet signálu, např.

.FOUR 1.1k V(1)

Výsledek: ve spektru se objeví „falešné“ vyšší harmonické.

82)

Namodelujte součet sinusovky o amplitudě 3V a kmitočtu 1kHz a kosinusovky o amplitudě 1V a kmitočtu 5kHz. Zobrazte v intervalu 0 až 5ms. Vypočítejte pomocí SPICE prvních 9 harmonických. Amplitudové spektrum zobrazte pomocí PROBE.

Řešení:

Zdroj sinus+cosinus

.param pi=3.14159

Esincos 1 0 value={3*sin(2*pi*1k*time)+cos(5*2*pi*1k*time)}

.tran 1u 5m

```
.FOUR 1k V(1)
.probe
.end
```

Ve výstupním souboru si ověřte, že ve spektru přibyla složka na kmitočtu 5kHz o amplitudě 1V a počáteční fázi 90 stupňů (kosinus). Činitel zkreslení THD je 33.3 %.

Proč jsme jako opakovací kmitočet zadali opět 1kHz?

83)

Vygenerujte 5 opakovacích period obdélníkového signálu o úrovních 0V a 10V, šířce impulsu 0.2ms a opakovací periodě 1ms. Strmost náběžné a sestupné hrany 1ns. Vypočtěte prvních 20 harmonických. Zobrazte v PROBE.

Řešení:

```
Vobdel 1 0 PULSE 0 10 0 1n 1n 0.2m 1m
.tran 1u 5m
.FOUR 1k 20 V(1); počet harmonických je 20
.probe
.end
```

V PROBE se omezte na zobrazení spektra do cca 20kHz.

Náměty:

V časové analýze měňte simulační čas po násobcích opakovací periody, tj. 1ms, 2ms, 3ms, Jaký to bude mít vliv na spektrum zobrazované v PROBE? Jaký vliv to bude mít na přesnost výpočtu spektrálních čar?

84)

Zjistěte spektrum výstupního napětí oscilátoru v ustáleném stavu. Změřte činitel harmonického zkreslení výstupního napětí.

Pozn.: Oscilátor byl řešen v příkladech 60 a 61.

Řešení:

V první fázi se změří kmitočet oscilací v ustáleném stavu (970.58 Hz).

V druhé fázi se doplní příkaz pro Fourierovu analýzu.

Oscilator

```
M1 D G 0 0 M2N6661
```





```
Vbat bat 0 10V
```

```
R1 D 1 10k
```

```
R2 1 2 100k
```

```
R3 2 G 1meg
```


```
C1 1 0 28.4nF
```



```

C2 2 0 2.84nF
C3 G 0 284pF
R4 bat D 100
.TRAN 1n 20m 0 20u SKIPBP
.FOUR 970.58 V([D])
.PROBE
.lib
.end

```



DC COMPONENT = 3.212885E+00

HARMONIC NO	FREQUENCY (HZ)	FOURIER COMPONENT	NORMALIZED COMPONENT	PHASE (DEG)	NORMALIZED PHASE (DEG)
1	9.706E+02	2.800E+00	1.000E+00	-5.201E+01	0.000E+00
2	1.941E+03	2.782E-01	9.936E-02	1.614E+02	2.654E+02
3	2.912E+03	2.373E-01	8.474E-02	-1.552E+02	8.423E-01
4	3.882E+03	5.549E-02	1.982E-02	-1.085E+02	9.951E+01
5	4.853E+03	3.286E-02	1.173E-02	9.776E+01	3.578E+02
6	5.823E+03	4.315E-02	1.541E-02	1.435E+02	4.555E+02
7	6.794E+03	1.549E-02	5.531E-03	-1.719E+02	1.921E+02
8	7.765E+03	1.106E-02	3.951E-03	3.575E+01	4.518E+02
9	8.735E+03	1.671E-02	5.966E-03	8.008E+01	5.481E+02

TOTAL HARMONIC DISTORTION = 1.338015E+01 PERCENT

V PROBE se však spektrum zobrazí nesprávně, a to ze dvou důvodů:

1. V analyzačním okně není periodický signál v ustáleném stavu,
2. Simulační čas není celistvým násobkem opakovací periody.



Postupem, uvedeným v příkladu 61, zavedeme počáteční podmínky, vedoucí k okamžitému ustálenému stavu. Simulační čas nastavíme na pětinašobek opakovací periody, tj. na 5.1516ms.

Oscillator

```

M1 D G 0 0 M2N6661
Vbat bat 0 10V
R1 D 1 10k
R2 1 2 100k
R3 2 G 1meg
C1 1 0 28.4nF
C2 2 0 2.84nF
C3 G 0 284pF
R4 bat D 100
.IC
+ V(1) = 2.0676387445
+ V(2) = 3.1072366239
+ V(D) = 1.0196013703
+ V(G) = 3.4454667233
+ V(bat) = 10.0000000000
.TRAN 1n 5.1516m 0 20u SKIPBP
.PROBE
.lib
.end

```



Poznámka – jiné řešení (jednodušší): přechodný stav „odřežeme“ nastavením parametru Tstart v příkazu .TRAN, například:

```

.TRAN 1n 20m 14.8484m 20u SKIPB

```

Zobrazí se přímo posledních 5 period ustáleného stavu.

85)

Zjistěte činitel harmonického zkreslení na výstupu zesilovače při jeho testování harmonickým napětím o kmitočtu 1kHz a amplitudě 5mV.

Řešení:

zesilovac

.param pi=3.14159

Ein vstup 0 value={5m*sin(2*pi*1k*time)}

Cv vstup baze 5u

Vbat bat 0 12V

Rb bat baze 757k

Rc bat vystup 2k

Q1 vystup baze 0 BC107A




.TRAN 1n 5m 1m 1u

.FOUR 1k V([vstup]) V([vystup])

.PROBE v([vstup]) v([vystup])

.lib

.end



Zkreslení výstupního signálu:

TOTAL HARMONIC DISTORTION = 4.392710E+00 PERCENT

Počáteční úsek 1ms je v časové analýze vynechán pro vyloučení případných přechodných dějů.

Zkreslení výstupního signálu již je patrné pouhým okem z časového průběhu (kladné „půlvlny“ jsou širší než záporné). Ze spektra je vidět, že za to mohou zejména dominantní 2. a 3. harmonická.

86)

Zobrazte spektrum amplitud výstupního napětí zesilovače, působí-li na jeho vstupu součet harmonických napětí, každé o amplitudě 5mV a kmitočtech 1kHz a 10kHz. Pomocí PROBE prostudujte intermodulační zkreslení zesilovače.

Řešení:

zesilovac

.param pi=3.14159

Ein vstup 0 value=

{5m*sin(2*pi*1k*time)

+5m*sin(2*pi*10k*time)}

Cv vstup baze 5u

Vbat bat 0 12V

Rb bat baze 757k

Rc bat vystup 2k



Q1 vystup baze 0 BC107A

.TRAN 1n 5m 1m 1u

.PROBE v([vstup]) v([vystup])

.lib

.end





Poznámky:

První 1ms je v časové analýze vynechána pro vyloučení případných přechodných dějů.

Ve spektru vstupního signálu jsou 2 dominantní spektrální čáry na kmitočtech 1kHz a 10kHz. Další čára na kmitočtu 14kHz je o 4 dekády menší a představuje numerickou chybu algoritmu FFT.

Ve spektru výstupního signálu jsou patrné spektrální čáry na tzv. kombinačních kmitočtech

$$mF1 \pm nF2.$$



87)

Zjistěte spektrální hustotu šumového výkonu a šumového napětí na kolektoru tranzistoru pro kmitočet 1kHz. Určete hustotu šumového výkonu jednotlivých zdrojů šumu v obvodu. Odhadněte efektivní hodnotu šumového napětí na výstupu zesilovače pro kmitočtové pásmo od 10Hz do 10kHz.


Pozn.: Zdroj proudu o nulovém proudu je pomocný, SPICE potřebuje označit zdroj pro přepočítání výstupního šumu na vstup.

Řešení:

tranzistor. zesil.

Vbat 3 0 5V

Rc 3 2 1k



Rb 3 1 120k

Ii 0 1 AC 1; střídavý proud může mít libovolnou hodnotu, nemá vliv na šumovou analýzu

Q 2 1 0 Q2N2221

.AC LIN 1 1k 1k

.NOISE V(2) Ii 1; jednička znamená, že výsledek šumové analýzy se zapíše do výstupního souboru .lib

.END

Výsledky lze nalézt ve výstupním souboru:

**** TRANSISTOR SQUARED NOISE VOLTAGES (SQ V/HZ)

Q1

RB	5.316E-20;...	tepelný šum, odpor báze
RC	1.488E-23;...	tepelný šum, odpor kolektoru
RE	0.000E+00;...	tepelný šum, odpor emitoru
IBSN	5.343E-14; ...	výstřelový šum, proud báze
IC	7.064E-16; ...	výstřelový šum, proud kolektoru
IBFN	0.000E+00;...	blikavý šum, proud báze
TOTAL	5.414E-14;...	celkový šum, dodávaný tranzistorem

**** RESISTOR SQUARED NOISE VOLTAGES (SQ V/HZ)

Rc Rb

TOTAL 1.560E-17 6.379E-16 ; ... tepelný šum, dodávaný odpory Rc a Rb

**** TOTAL OUTPUT NOISE VOLTAGE = 5.479E-14 SQ V/HZ
= 2.341E-07 V/RT HZ

TRANSFER FUNCTION VALUE:

V(2)/Ii = 6.795E+04

EQUIVALENT INPUT NOISE AT Ii = 3.445E-12 A/RT HZ

Veličiny SQUARED NOISE VOLTAGES jsou spektrální hustoty výkonu v [V^2/Hz].

Veličiny NOISE VOLTAGES jsou spektrální hustoty napětí v [$V/\sqrt{\text{Hz}}$].

Tranzistor „šumí“ více než vnější rezistory. Prakticky všechen šum tranzistoru pochází z výstřelkového šumu, jehož příčinou je klidový proud báze.

Z údaje TOTAL OUTPUT NOISE VOLTAGE = 2.341E-07 V/RT HZ

určíme efektivní hodnotu šumového napětí v kmitočtovém pásmu od 10Hz do 10kHz, tj. Δf je zhruba 10kHz:

$$U_{ef} = 2.341 \cdot 10^{-7} \sqrt{10000} = 23.41 \text{ mV}.$$

Pak napětí špička-špička bude cca 5 až 6 krát větší, tj. cca 120 μV .

Pozn.: Výpočet efektivní hodnoty platí za předpokladu, že spektrální hustota šumového výkonu je konstantní v celém audiopásmu 10Hz až 10kHz.

88)

Zjistěte kmitočtový průběh spektrální hustoty šumového výkonu na kolektoru tranzistoru v kmitočtovém pásmu 0.1Hz až 10MHz. Identifikujte příspěvky jednotlivých zdrojů šumu k celkovému šumu na výstupu.

Z křivky spektrální hustoty určete výkon šumu v kmitočtovém pásmu od 10Hz do 10kHz.

Určete kmitočtový průběh ekvivalentního vstupního šumu mezi bází a emitorem.

Řešení:

tranzistor. zesil.

Vbat 3 0 5V

Rc 3 2 1k

Rb 3 1 120k

Ii 0 1 AC 1

Q 2 1 0 Q2N2221

.AC DEC 10 0.1 10meg

.NOISE V(2) Ii; nepožadujeme tisk do výst. souboru

.probe

.lib

.END




Veškerá analýza dat ze SPICE se provede v PROBE.

Význam veličin V(ONoise), NTOT(ONoise), ... viz přednášky.

Největším přispivatelem k výstupnímu šumu je tranzistor, konkrétně výstřelový šum NSIB vyvolávaný klidovým proudem báze.

Integrací NTOT(ONoise), což je spektrální hustota šumového výkonu na výstupu, získáme výkon v daném kmitočtovém pásmu od 0,1Hz výše. Druhá odmocnina je pak efektivní hodnota napětí. Integrace se v PROBE provádí funkcí S (). Na kmitočtu 10kHz lze odečíst výkon 541.792pV^2 a efektivní hodnotu $23.405\mu\text{V}$. To je v dobré shodě s výsledky z příkladu 87 (tam jsme řešili šum od 10Hz, zde od 0.1Hz, rozdíly jsou prakticky nulové).

Z funkce INoise vyplývá, že výstupní šum si lze představit jako výsledek působení vstupního šumového proudu cca $3.4\text{pA}/\sqrt{\text{Hz}}$, který je následně zesílen na kolektor.






89)

Proveďte analýzu výstupního šumu zesilovače s uvažováním celého vstupního obvodu (zdroj střídavého signálu o vnitřním odporu 10 Ohmů a vazební kapacitor).

Poznámka:

Na nízkých kmitočtech, kdy zdroj signálu je od zesilovače oddělen vazebním kapacitorem, bude šumové napětí na výstupu stejné jako v příkladu 88.

Na vyšších kmitočtech bude k přechodu báze-emitor připojen vnitřní odpor zdroje. To bude mít dva následky: a) Šumové napětí tohoto odporu začne být zesilováno na výstup, b) budou ovlivněny příspěvky dalších zdrojů šumu na výstup.



Řešení:

tranzistor. zesil.

Vbat 3 0 5V

Rc 3 2 1k

Rb 3 1 120k

Vin x 0 AC 5mV

Ri x in 10

Cv in 1 5u

Q1 2 1 0 Q2N2221

.NOISE V(2) Vin

.AC DEC 10 0.1 10meg

.probe

.lib

.END

Celkový šumový výkon v pásmu do 10kHz poklesl na cca 34.7pV² a efektivní hodnota šumového napětí klesla na cca 5.89uV. Z křivek V(ONOISE) a NTOT(ONOISE) je zřejmý pokles šumové aktivity od kmitočtu cca 100Hz, kdy se začíná uplatňovat nízká reaktance vazebního kapacitoru. Vnitřní odpor zdroje se střídavě připojuje paralelně k přechodu báze-emitor a snižuje tak přenos výstřelového šumu v bazovém přechodu tranzistoru na výstup. Podstatný příspěvek k šumu tranzistoru pak představuje výstřelový šum kolektorového přechodu. Výstupní šum se v podstatě skládá z tohoto šumu a z příspěvku tepelného šumu vnitřního odporu zdroje.

90)

Analýzujte spektrální hustotu napětí na výstupu nízkošumového operačního zesilovače LT1113, zapojeného jako jednotkový zesilovač, v pásmu kmitočtů 0.1Hz až 100kHz.

Řešení:

sledovac s LT1113

Vin 1 0 AC 1

Rs 1 2 1

X1 2 3 4 5 3 LT1113/LT

Vplus 4 0 15V

Vminus 5 0 -15V



.lib

.AC DEC 10 0.1 100k

.NOISE V(3) Vin

.PROBE

.END



Do kmitočtu cca 100Hz se uplatňuje šum 1/F, pak bílý šum.

Na 1kHz vychází spektrální hustota šumového napětí cca $4.58\text{nV}/\sqrt{\text{Hz}}$.

91)

Pomocí vyhodnocovací analýzy (Performance Analysis) zjistěte závislost „desetikilohertzové“ spektrální hustoty výstupního šumového napětí na odporu Ri. Tento odpor krokujte logaritmičticky od 1 Ohmu do 100kOhmů.

Řešení:

sledovac s LT1113

Vin 1 0 AC 1

Rs 1 2 {Rs}

X1 2 3 4 5 3 LT1113/LT

Vplus 4 0 15V

Vminus 5 0 -15V

.lib

.param Rs 1



.step dec param Rs 1 100k 5

.AC DEC 10 0.1 100k

.NOISE V(3) Vin

.PROBE

.END



Použití vyhodnocovací analýzy v PROBE:

Trace/Performance Analysis/Wizard/Next

Vybereme měřicí funkci YatX

Next

Zvolíme křivku, na kterou aplikujeme měřicí funkci, tj. $V(\text{ONoise})$. Do políčka „X value to get Y value at“ vepíšeme 1k

Next

Next

Upravíme osu X na logaritmickou, uživatelské nastavení měřítka od 1 do 100k Ohmů.

Z grafu je zřejmé, že při odporu vnitřního zdroje do cca $1\text{k}\Omega$ je šum určen pouze vynikajícími šumovými vlastnostmi OZ. Při větších odporech pak začíná postupně převládat tepelný šum vnitřního odporu zdroje.

Příklady 92 až 100

Modelování teplotních závislostí součástek. Teplotní analýza. Příkaz .TEMP.

Modelování tolerancí součástek. Statistická analýza. Analýza .MC. Analýza .WCASE.

92)

U tranzistorového obvodu analyzujte závislost stejnosměrného napětí na kolektoru na teplotě od 0 do 50 stupňů Celsia. Uvažujte

a) teplotně nezávislé odpory RC a RB

b) odpor RB s lineárním teplotním koeficientem (0, 2000, 4000, 6000)ppm.

Řešení:

a)

tran. obvod

Vbat 3 0 5V

Rb 3 1 120k

Rc 3 2 1k



Q 2 1 0 Q2N2221

.DC temp 0 50 1

.probe

.lib

.end



Při vzrůstu teploty z 0 na 50 stupňů poklesne napětí kolektoru z 3.0051V na 2.3437V. Teplotní koeficient tohoto napětí je asi -13.2mV/°C. V obvodu není pracovní bod stabilizován.

b)

ppm znamená „part per milion“, neboli např. 2000ppm je $2000 \cdot 10^{-6} = 2\text{m/stupeň}$.

...

Rb 3 1 120k TC={TC}

Rc 3 2 1k


.param TC 1m

.step LIN param TC 0 6m 1m

...

Při teplotním koeficientu TC asi 5.4m bude napětí kolektoru téměř teplotně vykompenzováno.

Presvědčte se o tom, že teplotní závislost RC působí na V(2) opačným směrem než teplotní závislost Rb.



93)

U obvodu z příkladu 92 analyzujte teplotní závislost napětí kolektoru, jestliže:

a) teplota tranzistoru je stabilizována na 30 °C a odpory RB a RC jsou: jen RB teplotně závislý, jen RC teplotně závislý, oba teplotně závislé,

b) teplota tranzistoru je vždy o 15 °C vyšší než teplota odporů, oba odpory jsou teplotně závislé.

Řešení:

a)

Do modelu tranzistoru je třeba vpravit údaj o fixní teplotě tranzistoru, která bude nezávislá na globální teplotě:

$T_{ABS}=30$

Model tranzistoru je v originální knihovně na disku a není rozumné do ní zasahovat. Pak jsou 2 další možnosti:

Buď si okopírovat model do vstupního souboru a zde přidat údaj o fixní teplotě, nebo provést klonování originálního modelu syntaxí AKO (A Kind Of).

Použijeme druhou možnost.

tran. obvod

Vbat 3 0 5V

Rb 3 1 120k TC={TC}

Rc 3 2 1k; TC={TC}

.param TC 1m

.step LIN param TC 0 6m 1m

Q 2 1 0 Q2N2221x

.MODEL Q2N2221x AKO:Q2N2221 NPN T_ABS=30

;uvažujeme teplotní závislost RB

;teplotní závislost RC je nyní vyblokována

;zaveden nový model tranzistoru



tento nový model vznikl klonováním originálního modelu Q2N2221

.DC temp 0 50 1

.probe

.lib

.end



Z výsledků vyplývá, že teplotní závislost RB působí proti teplotní závislosti tranzistoru a teplotní závislost RC podporuje teplotní závislost tranzistoru.

b)


...

.MODEL Q2N2221x

+AKO:Q2N2221 NPN T_REL_GLOBAL=15

...

Zde parametr T_REL_GLOBAL znamená relativní zvýšení teploty součástky o 15 stupňů oproti globální teplotě.



94)

Namodelujte Band-Gap napěťovou referenci, která by na svých svorkách poskytovala napětí 1.25V při 25 °C. Teplotní závislost napětí je dána obrázkem, resp. vzorcem

$$V = a + b * TEMP + c * TEMP^2$$

kde $a=1.24859$, $b=8.75e-5$, $c=-1.25e-6$.

Řešení:

BANDGAP

.param a=1.24859 b=8.75e-5 c=-1.25e-6


Ebandgap 1 0 value={a+b*TEMP+c*TEMP^2}

.DC temp -55 125 1

.probe

.lib

.end

**95)**

Namodelujte rezistor, který by vykazoval teplotní závislost odporu podle lomené čáry na obrázku (viz výsledky simulace).

Řešení:

V SPICE lze teplotní závislost odporu modelovat přímo jen pomocí lineárního a kvadratického, resp. exponenciálního teplotního součinitele. Musíme proto použít nepřímé modelování, a to přes řízené zdroje.


Odpor R mezi svorkami x y lze modelovat zdrojem proudu řízeným napětím (G), tj. zdrojem o rovnici $I=GV$. Řídící svorky budou x a y a paralelně k nim budou výstupní svorky zdroje proudu. Pak mezi svorkami x a y naměříme odpor $R=1/G$. Vodivost G definujeme vzorcem, v němž může figurvat i teplota:

Gtemp x y value= {1/(vzorec pro odpor)}

Abychom mohli simulovaný odpor měřit, připojíme k němu zdroj napětí a v PROBE zobrazíme odpor jako

NAPĚTÍ_NA_ZDROJI/(-PROUD_ZDROJEM)

Mínus proto, že SPICE chápe směr proudu zdrojem od svorky + ke svorce – vnitřkem zdroje.



Vin 1 0 1V

G 1 0 value {1/TABLE(temp, 20,1k,70,5k)}; nastavte syntaxi funkce TABLE

.DC temp 0 100 1

.probe

.end

96)

Nalezněte histogram stejnosměrného napětí na kolektoru, jestliže odpory RC a RB mají tolerance 10% (Gaussovo pravděpodobnostní rozložení). Použijte 100 běhů analýzy Monte Carlo.

Řešení:

Tolerance odporů se definuje pomocí modelu odporu, konkrétně klíčovým slovem LOT nebo DEV u násobícího součinitele odporu R. Podrobnosti viz přednášky. Provedeme jednobodovou analýzu DC při pseudorozmítání napětí baterie Vbat. V příkazu .MC zadáme libovolnou povolenou mřicí funkci (např. YMax). Nakonec ji nepoužijeme, výsledky analýzy budeme zpracovávat v PROBE. Nesmíme zapomenout uvést OUTPUT ALL, aby data ze všech 100 simulačních běhů byla k dispozici pro PROBE.

Monte Carlo

Vbat 3 0 10V

Rc 3 2 odpor 1k

Rb 3 1 odpor 330k

Q 2 1 0 Q2N2222

.MODEL odpor RES R=1 lot=10%

.DC Vbat 10 10 1

.MC 100 DC V(2) YMAX OUTPUT ALL

.lib

.probe

.end

V PROBE se po analýze otevře okno vyhodnocovací analýzy. Přidáme křivku V(2). Objeví se histogram.

Desetiprocentní variace obou odporů způsobí nepatrné změny napětí na kolektoru, od 6.967V do 5.025V. Jak je to možné, když v obvodu není stabilizace pracovního bodu? Vysvětlení: LOT je tolerance, uplatňovaná při každém simulačním běhu na všechny součástky, napojené na model ODPOR, stejně. Tedy když se o 10 % změní RB, změní se o 10% i RC. Tyto změny se kompenzují (zvětšení RB zvýší napětí na kolektoru, zvětšení RC sníží napětí na kolektoru).

V praxi jsou tolerance RC a RB nekorelované. Individuálně pro každou součástku se nastavují slovem DEV.



...

.MODEL odpor RES R=1 dev=10%


...

Další histogram již odpovídá realitě. Střední hodnota napětí je 4.978V (mean), praktické hranice změn tohoto napětí jsou dány 10%ním a 90%ním kvantilem.

Při zvyšování počtu běhů MC se zřetelněji začne rýsovat Gaussova křivka (a v prvním případě tolerancí LOT půjde o rovnoměrné rozdělení).



n samples	= 100
n divisions	= 10
mean	= 4.99522
sigma	= 0.0168837
minimum	= 4.96734
10th %ile	= 4.97272
median	= 4.99457
90th %ile	= 5.01896
maximum	= 5.02479
3*sigma	= 0.0506512



n samples	= 100
n divisions	= 10
mean	= 4.97845
sigma	= 0.387142
minimum	= 4.05272
10th %ile	= 4.44437
median	= 5.01073
90th %ile	= 5.48226
maximum	= 5.78952
3*sigma	= 1.16143

Shrnutí:

DEV použijeme pro modelování nezávislých tolerancí součástek, které jsou napojeny na stejný model.

LOT použijeme na modelování vzájemně závislých tolerancí, např. tolerancí odporů CMOS na společném čipu.

Tolerance LOT a DEV lze kombinovat.

97)

U obvodu z příkladu 96 uvažujte navíc toleranci parametru BF tranzistoru 30%.

Řešení:

Toleranci libovolného parametru v modelu tranzistoru definujeme přidáním syntaxe LOT nebo DEV přímo za daný parametr do modelu. Abychom nemuseli zasahovat do originálního modelu, je možné opět využít techniky klonování modelu (viz příklad 93). K tomu ale budeme potřebovat znát velikost parametru BF. Toto lze zjistit například z výstupního souboru po výpočtu pracovního bodu (viz předchozí příklad), nebo přímo z modelu na disku. Výsledek je $BF=255.9$.

Monte Carlo

Vbat 3 0 10V

Rc 3 2 odpor 1k

Rb 3 1 odpor 330k

Q1 2 1 0 Q2N2222x

.model Q2N2222x

+AKO:Q2N2222 NPN BF=255.9 lot=30%

.MODEL odpor RES R=1 dev=10%


.DC Vbat 10 10 1

.MC 100 DC V(2) YMAX OUTPUT ALL

.lib

.probe

.end



n samples	= 100
n divisions	= 10
mean	= 4.97617
sigma	= 0.656692
minimum	= 3.06924
10th %ile	= 4.17257
median	= 4.87618
90th %ile	= 5.89599
maximum	= 6.16718
3*sigma	= 1.97008

Rozptyl klidového pracovního bodu se nyní podstatně zvětšil.


98)

U tranzistorového obvodu změřte vstupní impedanci (mezi svorkami 1-0) v kmitočtovém rozsahu od 1Hz do 100MHz. Uvažujte nekorelované tolerance odporů RB a RC 10% a toleranci BF tranzistoru 30%.

Pomocí vyhodnocovací analýzy získajte histograme vstupní impedance na kmitočtu 1kHz a mezního kmitočtu (kmitočtu třídecibellového poklesu impedance).

Řešení:

K bázi připojíme střídavý zdroj proudu. Jeho DC atribut je nulový, takže zdroj nebude narušovat nastavený pracovní bod. Nastavíme-li jeho atribut AC na jedničku, pak střídavé napětí na bázi bude číselně rovno vstupní impedanci.



```

Monte Carlo
Vbat 3 0 10V
Iin 0 1 AC 1
Rc 3 2 odpor 1k
Rb 3 1 odpor 330k
Q1 2 1 0 Q2N2222x
.model Q2N2222x
+AKO:Q2N2222 NPN BF=255.9 lot=30%
.MODEL odpor RES R=1 dev=10%
.AC DEC 10 1 100meg
.MC 100 AC V(2) YMAX OUTPUT ALL
.lib
.probe
.end


```

Postup v PROBE:

Histogramy přidáme pomocí performance Analysis“.

Měřicí funkce YatX zjistí souřadnici Y křivky, je-li zadána X-ová souřadnice. Pro 1kHz tak změříme vstupní impedanci na tomto kmitočtu. Měřicí funkce

Cutoff_Lowpass_3dB pak přímo změří kmitočet třicidbelového poklesu impedance.



99)


U tranzistorového obvodu zjistěte metodou Worst Case, jaká bude největší odchylka vstupní impedance na kmitočtu 1kHz od jmenovité hodnoty při uvažování tolerancí RB, RC a BF tranzistoru z příkladu 98.

Řešení:

```

Worst Case
Vbat 3 0 10V
Iin 0 1 AC 1
Rc 3 2 odpor 1k
Rb 3 1 odpor 330k
Q1 2 1 0 Q2N2222x
.model Q2N2222x
+AKO:Q2N2222 NPN BF=255.9 lot=30%
.MODEL odpor RES R=1 dev=10%
.AC lin 1 1k 1k ; aktivace jednobodové AC analýzy na kmitočtu 1kHz
.WCASE AC V(2) YMAX; aktivace analýzy Worst Case s měřicí funkcí YMAX
.lib
.end

```



Vysvětlení (podrobnosti viz přednášky):

YMAX je funkce, hledající v každém běhu absolutní hodnotu z maximálního rozdílu mezi aktuální a nominální křivkou. V případě jednobodové analýzy tato funkce vrací absolutní hodnotu rozdílu mezi hodnotou impedance v daném běhu a hodnotou nominální.

Program nejprve provede tolik běhů, kolik je parametrů se zadanou tolerancí. V každém běhu je jeden z parametrů zvětšen nad nominální hodnotu o zadaná procenta. Je zaznamenáno, zda došlo

k zvýšení či snížení sledované veličiny $V(2)$. Když jsou známy všechny tyto citlivosti, provede se finální analýza s dosazením parametrů na hranicích tolerančního pásma, buď horní nebo spodní hranice tak, aby se počítaly nejnepríznivější případy.

Výsledky z výstupního souboru (zkráceně):

Mean Deviation = 96.5

Sigma = 58.886

RUN *MAX DEVIATION FROM NOMINAL*

Rc ODPOR R 156.02 (2.65 sigma) higher at $F = 1.0000E+03$
(.8797% change per 1% change in Model Parameter)

Q2N2222X BF DEVICES 117.19 (1.99 sigma) higher at $F = 1.0000E+03$
(.6608% change per 1% change in Model Parameter)

Rb ODPOR R 16.297 (.28 sigma) higher at $F = 1.0000E+03$
(.0919% change per 1% change in Model Parameter)

WORST CASE ALL DEVICES

Device	MODEL	PARAMETER	NEW VALUE
Q1	Q2N2222x	BF	332.67 (Increased)
Rc	odpor	R	1.1 (Increased)
Rb	odpor	R	1.1 (Increased)

**** *SORTED DEVIATIONS OF V(2)* *TEMPERATURE = 27.000 DEG C*
WORST CASE SUMMARY

Mean Deviation = 52.3470E+03

Sigma = 0

RUN *MAX DEVIATION FROM NOMINAL*


WORST CASE ALL DEVICES

52.3470E+03 higher at $F = 1.0000E+03$
(129.52% of Nominal)

Závěr analýzy: K největší odchylce od nominální hodnoty impedance dojde o 29.5 % směrem nahoru, jestliže se vychýlí R_C , R_B i BF směrem nahoru o 10%, 10% a 30%.

100)

U tranzistorového obvodu zjistěte metodou Worst Case nejhorší případ, který může nastat pro změnu stejnosměrného pracovního bodu, konkrétně napětí na kolektoru, při uvažování tolerancí R_B , R_C a BF z příkladu 99.



Řešení:

Worst Case

Vbat 3 0 10V

Rc 3 2 odpor 1k

Rb 3 1 odpor 330k

Q1 2 1 0 Q2N2222x

.model Q2N2222x

+AKO:Q2N2222 NPN BF=255.9 lot=30%

.MODEL odpor RES R=1 dev=10%

.DC Vbat 10V 10V 1V; jednobodová DC analýza

.WCASE DC V(2) YMAX

.lib

.end

Výsledky z výstupního souboru (zkráceně):

Mean Deviation = -1.0115E-03

Sigma = 4.2879E-03

RUN MAX DEVIATION FROM NOMINAL

Rb ODPOR R 5.0006E-03 (1.17 sigma) higher at Vbat = 10
(1.0009% change per 1% change in Model Parameter)

Rc ODPOR R 4.7035E-03 (1.10 sigma) lower at Vbat = 10
(.9415% change per 1% change in Model Parameter)

Q2N2222X BF DEVICES 3.3317E-03 (.78 sigma) lower at Vbat = 10
(.6669% change per 1% change in Model Parameter)

WORST CASE ALL DEVICES

Device	MODEL	PARAMETER	NEW VALUE
Q1	Q2N2222x	BF	179.13 (Decreased)
Rc	odpor	R	.9 (Decreased)
Rb	odpor	R	1.1 (Increased)

WORST CASE SUMMARY

Mean Deviation = 1.8006

Sigma = 0

RUN MAX DEVIATION FROM NOMINAL

WORST CASE ALL DEVICES

1.8006 higher at Vbat = 10
(136.04% of Nominal)

Shrnutí: Při snížení BF, snížení Rc a zvýšení Rb na hranice tolerancí dojde k nejhoršímu případu – k vzrůstu kolektorového napětí o 36 %.